
1.	Introduzione

1.1.	Generalita'

Un consistente vantaggio della filosofia GBASIC è quello di estendere ad ogni programmatore la possibilità di usare particolari risorse fisiche, che pur richiedano procedure complesse, senza conoscenza del loro modo inter�no di funzionare. 

In questa maniera, chiunque può integrare velocemente nel proprio programma funzionalità, magari sviluppate e messe a punto da altri, con l'uso di "macro strumenti", all'uopo predisposti.

In questo manuale è descritto il driver G-LMD (GBASIC Lan Master Driver). Con esso si può estendere il GBASIC standard (v.7.31 o successiva) in modo da poter operare come master di una rete G-LAN. 

Esiste un corrispondente strumento software per le stazioni slaves de�nominato G-LSD, descritto nell'apposito manuale.

In una rete G-LAN, un solo sistema può disporre del driver master G-LMD mentre tutti gli altri, gli slaves, devono disporre di G-LSD.

1.2.	Prerequisiti

Il protocollo G-LAN è definito dal documento "G/DLAP - GBASIC-LAN Data Link Access Protocol" che si riterrà essere a perfetta conoscenza del lettore. Nel seguito, per brevità, detto documento sarà riferito come [1].

Si riterrà inoltre acquisita la conoscenza del GBASIC e dei concetti il�lustrati nel manuale "GBASIC v.7.30 - Manuale di Riferimento", nel seguito indicato come [2].

Le presenti note sono relative alla versione 7.31 di GBASIC e 1.00 di G-LMD. 

1.3.	Compatibilita'

Il protocollo G-LANè stato interpretato fondamentalmente come vei�colo di informazioni. Il driver, pertanto, non analizza nè interpreta in alcun modo il contenuto dei messaggi ma si limita a rendere possibile il loro tras�ferimento od a prendere nozione di condizioni anomale che determinino l'impossibilità di eseguirlo. 

L'unica eccezione alla norma sopra enunciata è costituita dai messaggi sotto elencati che sono riservati al sistema:

P - da master a slave

N - da slave a master

x - da master a slave

L'uso di qualsiasi altro tipo di messaggio è libero.

G-LMD gestisce 32 unità G-LAN identificate da 0 a 31.

G-LSD è compatibile con il driver master G-LMD e con GBASIC ver�sione 7.3x e 7.4x.

2.	Modo di uso

2.1.	Generalita'

Nel realiszzare GPM-D è stato posta particolare attenzione a conciliare una oggettiva, intrinseca, complessità della funzione richiesta con la facilità di uso e ad armonizzare logicamente la sua filosofia operativa con i criteri generali degli altri drivers della famiglia GBASIC, anche al fine di consentire in seguito una ulteriore evoluzione del driver in modo indipendente o quasi dai programmi ad alto livello.

Il metodo realizzato raggiunge un importante obiettivo: il programma�tore non deve considerare la rete G-LAN come di tipo Master / Slave ma ha l'impressione di avere a disposizione una linea di comunicazione dall'architettura paritetica dove poter inviare interrogazioni e recepire le relative risposte ma anche ricevere messaggi che apparentemente vengono inviati di propria iniziativa dai corrispondenti.

2.2.	Strumenti software

GBASIC e G-LMD mettono a disposizione i seguenti strumenti soft�ware:

1) la frase WRITE che viene usata per inviare messaggi sulla rete. Il parametro che indica il numero del record viene usato per indicare il numero della stazione destinataria.

2) La frase READ è impiegata per la lettura dell'esito della trasmissione di un messaggio. Le risposte possibili includono "*n" (timeout) e "$n" (conferma messaggio arrivato a destinazione).

3) la frase POP è adibita alla acquisizione di messaggi inviati dai corrispon�denti. Quando "arriva" un messaggio, viene generata una interruzione. La routine di interruzione utilizza la POP per prelevare il messaggio dall'apposito buffer FIFO. L'uso del FIFO dati, parallelo a quello dell'interrupt, garantisce la sicurezza di non "perdere" messaggi anche nel caso in cui questi pervengano a velocità di punta maggiore di quella consen�tita per il loro processo. 

4) la funzione DRVST() ritorna lo stato delle singole stazioni slaves (ONLINE/OFFLINE). Sono necessari due argomenti: uno che indica in modo consueto il numero del driver ed uno che indica la stazione. Il kernel, dalla vers. 7.3, è stato quindi modificato (i precedenti prevedevano un solo ar�gomento) pur conservando la compatibilità con i drivers precedenti. 

2.3.	Principio di funzionamento

Quando opera G-LMD sono attivi almeno tre processi indipendenti: 

1) l'esecuzione del programma GBASIC;

2) un processo detto TIMER che provvede in modo trasparente alla in�terrogazione delle stazioni slaves e ad una prima analisi delle risposte (il nome TIMER indica semplicemente la sua ciclicità e non deve indurre in con�fusione con i timers del GBASIC); 

3) un processo detto RECEIVER che viene attivato alla ricezione di ogni singolo carattere dalla linea seriale della rete. 

Il processo TIMER provvede ad interrogare ciclicamente le stazioni. In assenza di richieste dal GBASIC (dette "ad alto livello"), viene utilizzata una interrogazione di semplice polling cui le stazioni rispondono (di solito) con "nulla da dire". 

E' possibile che una o più stazioni non rispondano all'interrogazione di polling, ad esempio perchè inattive. Esperito un certo numero di tentativi, G-LMD deduce che le medesime sono inattive ed aggiorna una propria lista nella quale sono indicate le stazioni attive (ONLINE) o quelle inattive (OFFLINE). Le stazioni offline sono dette anche in "timeout". 

La lista sopra descritta viene utilizzata da G-LMD per stabilire la se�quenza di interrogazione: le stazioni inattive, infatti, vengono interrogate ad una frequenza assai più bassa rispetto a quelle attive, per accelerare il ciclo; è infatti presumibile che una stazione inattiva rimanga di solito nello stesso stato anche alla successiva interrogazione. 

Una variazione nello stato di una stazione, ossia un "risveglio" od una "scomparsa", determina però una condizione che il basso livello, ossia il driver G-LMD, non è in grado di gestire da solo. 

In questo caso, il driver "crea" un messaggio e lo invia all'alto livello come se questo fosse stato inviato dalla stazione.

Il messaggio può essere uno dei due seguenti: 

a) in caso di passaggio OFF->ON -  il messaggio "Cnn" dove nn è il numero della stazione che è passata in stato di ON.

b) in caso di passaggio ON->OFF - il messaggio "*nn" dove nn è il nu�mero della stazione che è passata in stato di OFF. 

La routine di servizio interrupt GBASIC provvede quindi POP- are la stringa dal FIFO ed a comportarsi di conseguenza. 

Se si invia un messaggio ad una stazione che risulta già trovarsi in timeout, G-LMD non esegue una effettiva interrogazione ma ritorna diretta�mente un messaggio di timeout.

2.4.	Accesso a G-LMD

G-LMD usa la funzione MMT (Mini Multi Task) del kernel. Per questo motivo è necessario procedere all'assegnazione di un CTC al compito di task switcher.

Esempio



open osc 3,"C 1 D 193"

osc 3, 40 for switch

Prima di aprire il driver è opportuno procedere subito ad avviare il baud rate generator per PumaLan, sempre a mezzo di un CTC, a meno che il clock non sia fornito per altra via.

Esempio



open osc 2,"C 2 D 194"

osc 2, 3

L'apertura del driver avviene poi in modo consueto con la frase OPEN e stringa di configurazione. Il nome del driver è "PUMAM:..". Dopo l'apertura è necessario assegnare al driver un livello di interrupt.

Esempio



open ch 3, "GLANM:..","S 3 D 178 C 179 V 179" 

ch 3 to interrupt 2

on interrupt 2 gosub ListenLan

Dopo l'apertura, è necessario procedere a due ulteriori operazioni per rendere il driver attivo: l'inizializzazione cold start e l'avvio.

La prima inizializza la tabella con i numeri di messaggio (per l'uso dei numeri di messaggio vedi [1]). Serve solo per sistemi muniti di RAM "continua", ossia con batteria tampone. Viene comandata con un messaggio speciale come segue:



write ch 3, 0, "!"

L'inizializzazione della tabella dei numeri di messaggio, in questo caso, avviene solo una volta. Ad ogni riaccensione, il driver sa con esattezza quale numero di messaggio inviare e quale aspettarsi dai corrispondenti e consente quindi di  conoscere l'esito di una trasmissione anche nel caso di mancanze di alimentazione.

Se il master o uno degli slaves non dispongono di RAM non volatile si utilizzerà la frase di cui sopra ad ogni inizializzazione.

Qualsiasi funzione di rete non diviene attiva fino al momento in cui non venga eseguita l'operazione detta di "startup" per mezzo della frase



write ch 3, <exp>, "+"

Questo ha lo scopo di consentire la completa predisposizione del sistema (in particolare l'interrupt) prima di avviare la rete che di solito all'inizio determina l'arrivo di numerosi messaggi.

Al parametro <exp> può essere  assegnato il valore da utilizzarsi per il numero di tentativi ammessi in caso di errori di comunicazione. Esso influ�enza il numero massimo di ritrasmissioni dello stesso messaggio ed il nu�mero massimo di richieste di ritrasmissione allo slave. Se <exp> è posto uguale a zero viene usato il valore di default 4 (4 ritrasmissioni + 4 richieste di ripetizione). Valori più bassi accelerano il ciclo ma rendono più probabile il timeout. Valori più alti rallentano il ciclo ma garantiscono una maggiore probabilità di trasferimento del messaggio.

Il programma GBASIC può interrogare una qualunque stazione con la frase WRITE, indicando il numero della stazione stessa come numero di re�cord.

Esempio



WRITE CH 3,6,"S0" 

invia la stringa S0 alla stazione n.6. 

Il programma, subito dopo effettuata la WRITE, esegue di norma una READ (in questo caso il numero di stazione non viene controllato) per ac�quisire la risposta proveniente dalla stazione. In ogni caso una risposta per�verrà certamente: o la risposta di conferma "$nn" o "*nn" che indica che nes�suna risposta è pervenuta.

Esempio



read ch 3,0,a$

La lettura dopo la WRITE è obbligatoria. Se si tentano due scritture consecutive senza avere operato la lettura, si ha un messaggio di errore (DRV ERR).

Una interrogazione eseguita da GBASIC non determina quindi una successiva interruzione; come già accennato le interruzioni conseguono ad eventi imprevedibili, come un timeout o un messaggio da un corrispondente.

La funzione DRVST(n,m) consente infine di conoscere in ogni istante lo stato di una certa stazione m (0=OFFLINE, 1=ONLINE), essendo n il nu�mero del canale associato al driver.

Esempio riepilogativo

Il sottostante programma (provato in pratica) dimostra le funzionalità di base del driver. Periodicamente esegue interrogazioni programmate dello stato di una stazione mentre nel frattempo stampa i messaggi provenienti dalle stazioni.



'       * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



'       *     PROGRAMMA DI PROVA MASTER RETE



'       *     VERSIONE 1.00



'       * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



'		Data:7/6/92



#DECLCONST



'--------------------------------------------------------------------------

'	Costanti varie

'--------------------------------------------------------------------------

	

	VERSIONE $ "STAMP"	'STAMP e' parola riservata per ora e data



	SI	= "1"

	NO	= "0"

	

	FOROUT = "0"

	

	FIFO	= "10"		'usata in info()



	TESTN	= "0"



'--------------------------------------------------------------------------

'	Canali

'--------------------------------------------------------------------------



	CONSOLE	= "0"

	GLAN		= "1"

	

	INKEY		= "4"

	RAMFILE 	= "5"



'--------------------------------------------------------------------------

'	Storages

'--------------------------------------------------------------------------



#END



#DECLVAR

	a,b,c,i		'uso generale fuori int

	d,e,f			'uso generale dentro int

	Dest

	

'	stringhe



	FromLan		'da LAN

	FrP			'tempor. per precedente

	Esito			'esito trasmissione LAN

	

#END		



#CODE



	goto	Avvio



'--------------------------------------------------------------------------

'		     ROUTINES DI SERVIZIO INTERRUPT

'--------------------------------------------------------------------------



'--------------------------------------------------------------------------

'	ListenLan - servizio messaggi arrivati da rete

'--------------------------------------------------------------------------



#-> ListenLan

	pop FromLan$

	print FromLan$			'invia risposta intatta ad host

	return interrupt 2



'--------------------------------------------------------------------------

'	Avvio - inizio effettivo del programma

'--------------------------------------------------------------------------



#-> Avvio



	gosub IniSys			'inizializza sistema

	

	print "SLM 1.0"

	print "Compilazione ";VERSIONE

	



'--------------------------------------------------------------------------

'	attesa di una stringa da host

'--------------------------------------------------------------------------



#-> UsP2

	print

	let b$=""			'b$ e' la stringa dove viene accumulato

	let a$=""			'il messaggio

#-> Loop

	input ldev INKEY,a$

	if a$="\" then SndStg	'accumula fino all'arrivo di CR

	if a$="!" then UsP2

	if len(a$)=0 then Loop

	let b$=concat$(b$,a$)

	goto Loop

	

'--------------------------------------------------------------------------

'	invio stringa in rete

'--------------------------------------------------------------------------



#-> SndStg

	let a$=substr$(b$,1,2)

	

'	se la stringa non inizia per numero si sottintende il destinatario

'	specificato in precedenza

	

	if val(a$)<>-1 then 

		let Dest = val(a$)			'destinatario

		let i=len(b$)-2			'lunghezza del messaggio

		let b$=substr$(b$,3,i)		'escludi i primi due caratteri

	endif

	

	print "Invio..";b$;" a";Dest

	write ch GLAN, Dest, b$

	read ch GLAN, 0, Esito$



'	processare qui situazioni di timeout



	print "Esito..";Esito$

	goto UsP2



	

'--------------------------------------------------------------------------

'	IniSys - inizializza sistema all'accensione

'--------------------------------------------------------------------------



#-> IniSys





'	------- dimensionamenti



	dim a$(80),b$(80),c$(80),d$(80), FromLan$(80), Esito$(20)



	reset interrupt	clear		'azzera FIFOs



'	------- apertura drivers



'---------------------------------------------------------------------------

'	CTC / baud rate generators

'---------------------------------------------------------------------------





'	open osc 3,"C 0 D 36"		'baud rate generator lan

'	osc 3, 1				'se non esiste via hardware



	open osc 4,"C 2 D 38"		'real time clock per MMT

	osc 4, 40 for switch



'---------------------------------------------------------------------------

'	drivers vari

'---------------------------------------------------------------------------



	open ch GLAN, "GLANM:..", "S 1 D 70 C 71 V 71" 

	ch GLAN to interrupt 2

	on interrupt 2 gosub ListenLan



'	break NO			'per evitare break non voluti

	autostart SI



	interrupt SI

	write ch GLAN, 0, "!"

	write ch GLAN, 2, "+"

	return

	



#END



2.5.	Struttura dei messaggi

Utilizzando il protocollo G-LAN per nuove applicazioni, si suggeris�cono la seguente struttura di messaggio: 



Lettera chiave��Mittente (2 caratteri)��Destinatario (2 caratteri)��Altri dati ecc.��

Nel modo descritto, future versioni di G-LMD potranno provvedere anche alla ritrasmissione di messaggi tra due stazioni.

2.6.	Un esempio applicativo

Consideriamo un esempio di massima con situazioni abbastanza fre�quenti nel pratico impiego della rete G-LAN. 

Supponiamo che il master debba autorizzare delle transazioni e ricev�erne poi i dati per l'archiviazione. Come già accennato, la particolare or�ganizzazione di G-LMD consente, in buona misura, di trattare una rete che è di tipo master/slaves come una rete paritetica. E' infatti possibile inviare una interrogazione ad una stazione così come è possibile apparentemente l'arrivo asincrono di messaggi dalle stazioni (che in realtà sono interrogate ciclica�mente dal driver in modo invisibile all'utente). 

Nel seguito, pertanto, continueremo a parlare, anche se impropria�mente, di "messaggi in arrivo dalle stazioni" quando questo non possa essere fonte di confusioni. Il programma esempio  prevede una routine di gestione interruzioni che ha lo scopo di trattare i messaggi spediti dalle stazioni. Il normale ciclo è iniziato dallo slave che, a seguito di un evento locale, richiede l'autorizzazione ad eseguire la transazione. 

La routine di servizio interruzioni, riconosciuta la richiesta provvede ad inviare, se esistente, l'autorizzazione. 

Alla fine della erogazione, la stazione invia i dati che il master registra. 



	

	open ch GLAN, "GLAM:..", "..."

 	...........

	write ch GLAN, 0, "!"

	write ch GLAN, 0, "+" 			'startup lan

	.....

#->> IntServ

	pop FromLan$

	let KeyLett$ = substr$(FromLan$,1,1)	'lettera chiave

	let N$ = substr$(FromLan$,2,1)    	'n. di stazione

	let NSta = val N$

	if KeyLett$ = "A" then SendCred		'rich. credito

	if KeyLett$ = "B" then RecTrans		'dati della transazione

#->> IntServExi

	return interrupt 2



#->> SendCred

	write ch GLAN, NSta, "C"   		'autorizzazione alla transazi�one

	read ch GLAN, NSta, a$

	let KeyLett$ = substr$(a$,1,1)		'lettera chiave

	if KeyLett$ = "$" then IntServExi



 .. qui vengono processate altre condizioni per risposte diverse



#->> RecTrans

 .. esegue il controllo e la registrazione dell'erogazione

	write ch GLAN, NSta, "D"   		'cancella ultima transazione

	read ch GLAN, NSta, a$ 

	let KeyLett$ = substr$(a$,1,1)		'lettera chiave

	if KeyLett$ = "$" then IntServExi

 .. qui vengono processate altre condizioni per risposte diverse

In questo ultimo caso la cancellazione dei dati della transazione dalla memoria dello slave avviene in seguito ad un comando del master, per con�seguire una maggiore sicurezza. 

2.7.	Condizioni di errore

G-LMD rileva alcune condizioni di errore che sono passate al GBASIC sotto forma di codice. Future versioni dell'interprete potranno pertanto esplicitarle (al momento risulta solo il messaggio DRV ERR).



1 - numero di stazione errato

2 - record troppo lungo

3 - WRITE ripetuto prima arrivo risposta

4 - WRITE ripetuto prima del READ

5 - arrivata risposta inattesa (err.interno)

3.	Algoritmi

Confidenziale - per solo uso interno

3.1.	Funzionamento base

Come già accennato il driver G-LMD prevede due processi in�dipendenti che corrono parallelamente alla esecuzione del programma GBASIC: 

TIMER (PROCLAN nel sorgente)  - attivato a scadenze periodiche dallo MMT di GBASIC;

RECEIVER - attivato dalla ricezione dei caratteri sulla SIO della rete. 

Il processo TIMER si autoinstalla alla OPEN nella lista dei processi chiamati da MMT. 

In realtà il processo TIMER non viene chiamato ad ogni "tic" di MMT. Quello che viene chiamato è un processo di prescaling (PRETIMER) che chiama TIMER ogni numero di TIC prefissati (costante PRESC_K). PRETIMER ha la funzione di trasmettere i caratteri verso le stazioni, se il buffer relativo non è vuoto. Ad ogni TIC pertanto viene inviato un carattere. Dopo PRESC_K TIC viene chiamato TIMER. 

Ogni chiamata a TIMER o RECEIVER determina l'esecuzione di un de�terminato programma in funzione di una variabile detta di stato (rispettivamente TIMSTAT e RXSTAT). Gli stati possibili sono: 

Per TIMER:

IDLE - stato inoperoso - viene avviata la trasmissione di una interrogazione. Questa sarà di normale polling se il GBASIC non ha presentato richieste o l'interrogazione GBASIC in caso contrario. 

WAITING - attesa di risposta dalla stazione. Se la risposta non inizia a per�venire entro un tempo prefissato (dato dalla costante WAITMAX) si il pas�saggio nello stato di TIMEOUT e la ripeti�zione della interrogazione per un numero massimo volte fis�sato dalla variabile REPCNT inizializzata dal sistema al valore REPMAX e modificabile dall'utente con la WRITE dest, n, "+" (illustrata in precedenza). Se la stazione non risultava già in time�out, si ha la generazione di una interruzione al GBASIC per segnalare la condizi�one anomala. 

RECEIVED - il messaggio è stato ricevuto. Viene analizzato anche sulla base delle condizioni a lato (ad esempio, tabel�la delle stazioni in timeout) e  de�termina le azioni conseguenti. Se opportuno, viene generata una interruzione GBASIC.  

Per RECEIVER:

ST0 - attesa SOH o ACK o NAK 

ST1 - attesa byte alto indirizzo

ST1_2 - attesa byte basso indirizzo

ST1_3 - attensa numero di messaggio

ST2  - attesa STX

ST3 - attesa messaggio 

ST4 - attesa 1.o bcc (carattere di controllo)

ST5 - attesa 2.o bcc (carattere di controllo) - messaggio pronto attesa lettura

Il ciclo base del processo TIMER è il seguente:

idle -> waiting -> received -> idle...

Come già accennato, G-LMD mantiene una tabella con lo stato delle stazioni. Ad ogni nuova interrogazione automatica, viene prelevato dalla tabella il numero della prossima stazione da interrogare tenendo presente che solo ogni ONOFFR stazioni attive viene interrogata una stazione in time�out per i motivi di ottimizzazione a suo tempo esposti.

Si osservi una situazione particolare: supponiamo che tutte le stazioni siano in timeout (ad esempio perchè, con alimentazione comune, siano tutte spente) ed immaginiamo di attivarle tutte simultaneamente. Il risveglio della prima stazione sarà rilevato quasi istantaneamente. Ma la seconda stazione sarà rilevata solo dopo ONOFFR interrogazioni della stazione attiva e così via per le successive.

3.2.	Gli stati del processo TIMER

IDLE

La figura 1 riporta il diagramma a blocchi semplificato dell'algoritmo usato per lo stato IDLE. Niente è operativo fino a che non si esce dallo stato di startup. Qualora il buffer di trasmissione non sia ancora vuoto (per una precedente interrogazione) si abortisce istantaneamente qualsiasi ulteriore operazione. Si esamina quindi il flag PRE_BUF2 che, se settato, indica che il programma GBASIC ha predisposto una interrogazione da trasmettere. In caso negativo si determina il numero della prossima stazione da interrogare e si avvia l'invio di un messaggio di polling. Viceversa si invia il messaggio predisposto dal GBASIC alla stazione richiesta. 

�INCORPORA MSDraw   \* unisciformato���

Si noti l'ingresso secondario REIDLE che viene usato per ritrasmettere interrogazioni in caso di timeout o di NAK dalla stazione. In uscita si passa in stato di WAITING per attendere la risposta. 

Nella figura 2 le operazioni dello stato di IDLE sono riportate nel det�taglio. Senza soffermarsi su quanto già illustrato, si noti che prima di eseguire il controllo di PRE_BUF2 si pongono le variabili WAITCNT e REPCNT e TORCNT al loro valore di default. Il loro scopo è di limitare la massima permanenza nello stato di WAITING (WAINTCNT) nonchè il nu�mero massimo di ripetizioni dell'interrogazione (REPCNT). 

Seguendo adesso il ramo "AUTO" delle interrogazioni automatiche, si può vedere come la variabile ONCOUNT sia usata per determinare il rappor�to tra le interrogazikoni a stazioni ON o OFF. Quando ONCOUNT, decre�mentato, raggiunge 0 è tempo di interrogare una stazione OFF. 

Nel caso in cui si debba interrogare una stazione ON si chiama la rou�tine NEXTON che ritorna il numero della prossima stazione ON. Se non ce ne sono si passa necessariamente ad interrogare una stazione in OFF. Viene quindi posta la variabile LASTON pari al numero della pompa in ON. Parimente si predispone, nel buffer di trasmissione, il campo indirizzo uguale al numero della stazione da indirizzare. Infine si pone il flag WAKEUP uguale a ON. Questo flag è usato per determinare lo stato noto della stazione prima della interrogazione per rilevare i "risvegli" e le "scomparse". Se la stazione risultava in ON, un timeout implicherà la neces�sità di generare un'interruzione al GBASIC per comunicare la "scomparsa". Viceversa per comunicare un "risveglio". L'operazione nel caso di stazione in OFF è la duale di quella sopra descritta.

La tabella interna delle stazioni conserva anche il numero di messag�gio da trasmettere ed il numero di messaggio che ci attende in risposta. Essi vengono prelevati per inserire il numero dell'interrogazione in partenza nel buffer di trasmissione, e quello che ci si aspetta in risposta in EXPNMSG (esso è previsto per usi futuri ma non viene sfruttato al momento). I numeri di messaggi in tabella sono incrementati solo quando si è certi che le interro�gazioni sono state recepite dallo slave.

Viene quindi settata la variabile PDEST che contiene il numero della pompa di destinazione e che deve essere settata tanto per le interrogazioni automatiche che per quelle GBASIC. 

Tornando indietro e seguendo il ramo "STARTGB" si osserva che ven�gono svolte operazioni preliminari del tutto analoghe a quelle sopra de�scritte. 

Interrogazioni GBASIC a stazioni in timeout ritornano direttamente un messaggio "*nn".

�INCORPORA MSDraw   \* unisciformato��� 



E' da illustrare tuttavia l'uso che viene fatto della variabile BUF2. Essa indica, se diversa da 0, che una interrogazione GBASIC è già in corso e che altre non possono essere effettuate. Quando una nuova richiesta viene inseri�ta dal GBASIC nel buffer di trasmissione, si verifica che il flag BUF2 sia OFF; in caso contrario viene segnalato un errore. L'inserimento della nuova inter�rogazione GBASIC determina, come si è visto sopra, il settaggio di PRE_BUF2.

WAITING

Ben più semplice è lo stato WAITING in cui non avvengono altre op�erazioni se non il controllo della variabile WAITCNT già descritta e le even�tuali ripetizioni.

RECEIVED

Come accennato, lo stato RECEIVED (fig.4) provvede ad una seconda analisi del messaggio ricevuto (la prima avviene direttamente a livello di processo RECEIVER). In primo luogo si verifica che non siano intervenuti er�rori di comunicazione (di parità o di bcc). In questo caso si trasmette un NAK e si riabilita il processo RECEIVER, contando il numero di ritentativi con la variabile NAKCNT inizializzata in precedenza al valore di default e si torna in stato di WAITING. Se il numero di tentativi supera il massimo, si procede come sotto descritto per il timeout.

Se non erano stati rilevati errori, si esamina il primo carattere del buffer. Si hanno le seguenti possibilità: 

SOH - inizio di un messaggio di risposta - si incrementano i numeri di mes�saggio - si invia ACK - si riabilita RECEIVER - si passa a stato READY. 

NAK - entra in stato di WAITING passando però attraverso REPEAT.

ACK - incrementa i numeri di messaggio - copia il messaggio di ACK "$n" nel buffer di ricezione deducendo n (numero di stazione) da PDEST (numero della sta�zione interrogata). 

Altro carattere - ci si comporta come con NAK.

Un messaggio è arrivato; se la stazione era in precedenza classificata tra quelle quiescenti, si deve aggiornare la tabella e metterla come presente. Se si la risposta è arrivata in seguito ad una interroga zione automatica, si deve generare una interruzione per segnalare il nuovo stato. Viceversa, se l'interrogazione è stata fatta dal GBASIC si copia il messaggio ricevuto nella apposita area.

In caso di interrogazione automatica di una pompa già considerata online, si esamina la risposta. Se questa ha il carattere chiave "C" ("nulla da dire"), si ritorna nello stato IDLE avviando un nuovo ciclo. Le risposte "nulla da dire", quindi, in questo caso, non provocano alcuna azione da parte del driver. 

Se, in caso di interrogazione auto ad una pompa già considerata attiva, il messaggio di risposta è diverso da "C", si esegue un push del messaggio nel FIFO dati e si richiede un interrupt al GBASIC. Similmente in caso di ris�veglio conseguente ad una interrogazione automatica. 

La variabile REPCNT tiene conto del numero delle ripetizioni e le limita. Quando infatti REPCNT raggiunge il valore zero si esegue la proce�dura RETTMO che ha lo scopo di prendere le oppor tune decisioni in seguito al timeout verificato. Se la richiesta era stata fatta in modo automatico, ossia non dal GBASIC, si controlla se la stazione risultava già in timeout. In questo caso non si esegue alcuna particolare operazione ma attraverso SHORTRDY si torna in stato IDLE. Se l'interrogazione era del GBASIC o se la stazione non risultava in timeout, si inserisce il messaggio di timeout nel buffer ("*n") ma nel primo caso non si fa interrupt e nel secondo invece si esegue push ed int. 

3.3.	Gli stati del processo RECEIVER

Gli stati del processo RECEIVER sono illustrati nel diagramma di figu�ra 5. La semplicità concettuale e gli abbondanti commenti nel listato non richiedono ulteriori informazioni. 

Il listato effettivi della versione in corso può differire da quello sotto riportato che ha solo scopo di riferimento.



;============================================================================

;

;	DRIVER PER RETE LOCALE "G-LAN"

;

;	GBASIC COMPATIBILE

;

;	VERS. 1.00 9/91

;

;	USABILE PER QUALSIASI SIO ATTRAVERSO STRINGA DI CONFIGURAZIONE

;

;	COMPLETAMENTE CONFIGURABILE DA SORGENTE

;

;============================================================================





	CSEG



	INCLUDE B7.CNF



;----------------------------------------------------------------------------

;	EQUATES

;----------------------------------------------------------------------------



;----------------------------------------------------------------------------

;	PARAMETRI MODIFICABILI IN FASE DI ASSEMBLAGGIO

;----------------------------------------------------------------------------



;------	NUMERO MASSIMO DI STAZIONI COLLEGABILI (BINARIO)



SLAVMAX	EQU	31		;NUMERO MASSIMO DI STAZIONI COLLEGABILI 



;------ RAPPORTO TRA CHIAMATE A MMT (GBASIC) E CHIAMATE AL PROCESSO PROCLAN

;	DETERMINA LA FREQUENZA DELLE INTERROGAZIONI ED IN GENERE TUTTE 

;	LE ATTESE



PRESC_K	EQU	8		;ITERAZIONI PRESCALER PRIMA DI CHIAMARE PROCLAN



;------	MASSIMA PERMANENZA IN STATO WAITING - DETERMINA IL TEMPO MASSIMO

;	DI ATTESA NEL CASO DI MANCATA RISPOSTA - NON LIMITA LA DURATA DELLA

;	RISPOSTA, POICHE' IL TEMPO SI 'AUTOALLUNGA' SE VENGONO RICEVUTI

;	DEI CARATTERI



WAITMAX EQU	5		;MASSIMA PERMANENZA IN STATO WAITING



;------	NUMERO MASSIMO DI RITRASMISSIONI E/O DI NAK DA PARTE DEL MASTER 

;	QUESTO E' IL VALORE DI DEFAULT ED E' MODIFICABILE DA PROGRAMMA

;	AL MOMENTO DELLA INIZIALIZZAZIONE DELLA RETE CON LA FRASE: 

;	WRITE DEST,RTY,"+"  - RTY E' IL NUOVO VALORE DI REPMAX



REPMAX	EQU	4		

				;E DI NAK CHE PUO' SPEDIRE MASTER (DEFAULT)



;------	RAPPORTO TRA INTERROGAZIONI A STAZIONI CHE RISULTANO ONLINE

;	E STAZIONI CHE RISULTANO OFFLINE - INFLUISCE SUL TEMPO DI ENTRATA

;	IN RETE DI UNA STAZIONE QUANDO VIENE ATTIVATA

 

ONOFFR	EQU	40		;NUMERO DI INTERROGAZIONI DI STAZIONI ON PRIMA

				;DI INTERROGARE UNA DELLE STAZIONI OFF



;------	LUNGHEZZA BUFFERS DI TRASMISSIONE E RICEZIONE



LANBUFL	EQU	80		;LUNGHEZZA BUFFERS LAN RX E TX

	

;------	FATTORE DIVISIONE CLOCK SIO - SETTARNE UNO SOLO A 1 E L'ALTRO A 0



CLK16	EQU	0		;SIO DIVIDE PER 16

CLK64	EQU	1		;SIO DIVIDE PER 64



;----------------------------------------------------------------------------

;	ALTRI PARAMETRI

;----------------------------------------------------------------------------



SYSTEM	EQU	5

			

ONLINE	EQU	1

OFFLINE	EQU	0



;	VALORI ASSUNTI DA RXERR



NOERR	EQU	0

BCCERR	EQU	1

PERR	EQU	2



;	VALORE ASSUNTO DA MREADY



;	(OMONIMI DEGLI STATI DI TIMSTAT)



;IDLE	 EQU	0		;PRONTO PER TRASMETTERE UNA DOMANDA

;WAITING EQU	1		;ESEGUITO WRITE, ATTESA DI RISPOSTA

;READY	 EQU	3		;RISPOSTA RICEVUTA - ATTESA DI READ



;	I VALORI ASSUNTI DA RXSTAT E DA TIMSTAT SONO DEFINITI NEL SEGUITO



	INCLUDE	CARATT.DEF



CR	EQU	0DH

SOH	EQU	1		;START OF HEADING



	IF	SLAN

PIOAD	EQU	4

PIOAC	EQU	6

	ENDIF

	

;----------------------------------------------------------------------------

;		     SEZIONE DI PUBBLICO ACCESSO

;----------------------------------------------------------------------------



HEADER:

;	DRIVER HEADING



	DB	1BH		;INDICA DRIVER PRESENTE



	DW	DRVEND-HEADER	





	DB	1		;DEFINIZIONE DEL TIPO - VALORI AMMESSI

				;0 - CHAR ORIENTED

				;1 - BLOCK ORIENTED

				;2 - SPECIAL DEVICE



	DB	'GLANM:..'	;NOME DEL DRIVER - SEMPRE 8 CARATTERI



;----------------------------------------------------------------------------

;	INIZIO DELLA PORZIONE ESEGUIBILE

;----------------------------------------------------------------------------



;	IN INGRESSO:

;	HL - INDIRIZZO DEL BUFFER

;	DE - NUMERO DEL RECORD IN BINARIO

;	B  - CODICE OPERAZIONE RICHIESTA

;		VALORI AMMESSI:

;		0 = SCRITTURA

;		1 = LETTURA

;		2 = INIZIALIZZAZIONE



;	CODICI DI RITORNO: (IN ACC)

;		0 - TUTTO OK

;		1 - CODICE DI OPERAZIONE RICHIESTO NON AMMESSO DA QUESTO

;		    DRIVER

;		2 - BUFFER PIENO



;	NOTA - QUESTO DRIVER OPERA SOLO IN OUT - I DATI LI RITORNA SOLO

;	ATTRAVERSO IL FIFO (POP)



;	DRIVER NON VIENE DICHIARATO PUBLIC - VIENE TROVATO AUTOMATICAMENTE

;	DAL SISTEMA E CHIAMATO POI DALLA SUB 'DRIVER' INTERNA AL GBASIC



DRIVER:	LD	A,B		;DECIDI QUALE OPERAZIONE ESEGUIRE

	

	CP	0

	JR	Z,DRV_WR

	

	CP	1

	JP	Z,DRV_RD

	

	CP	2

	JP	Z,INIT

	

	CP	6

	JP	Z,STATUS

	

	LD	A,1		;ERRORE - CODICE RICHIESTA NON AMMESSO

	RET

	

;----------------------------------------------------------------------------

;	FUNZIONE N.0 - SCRITTURA RECORD

;----------------------------------------------------------------------------



DRV_WR:	PUSH	HL		;INDIRIZZO BUFFER DA TX

	PUSH	DE		;NUMERO DEL RECORD IN BINARIO



;----------------------------------------------------------------------------

;	MESSAGGIO SPECIALE '+' - AVVIA RETE

;----------------------------------------------------------------------------



	LD	A,(HL)		;MESSAGGI DI STARTUP INIZIANO PER "+"

	CP	'+'		;IL RESTO E' IGNORATO

	JR	NZ,NOSTUP



	XOR	A		;FINE DELLA FASE DI STARTUP

	LD	(STARTUP),A

	

	LD	A,E		;IL NUMERO DI RECORD SE <> 0 INDICA NUM RETRY

	OR	A

	JR	Z,NOSETR

	LD	(MAXRTY),A

	

NOSETR:	POP	DE

	POP	HL

	XOR	A

	RET





;----------------------------------------------------------------------------

;	MESSAGGIO SPECIALE '-' - ARRESTA RETE

;----------------------------------------------------------------------------



NOSTUP:	CP	'-'		;MESSAGGIO DI STOPDOWN?

	JR	NZ,NOSTDN



	LD	A,1		;SOSPENDI ATTIVITA'

	LD	(STARTUP),A

	JR	NOSETR





;----------------------------------------------------------------------------

;	MESSAGGIO SPECIALE '!' - RESET GENERALE

;----------------------------------------------------------------------------



;	(DA ESEGUIRE ALL'INSTALLAZIONE)



NOSTDN:	CP	'!'		;MESSAGGIO DI RESET GENERALE?

	JR	NZ,NOGENRES



	PUSH	BC

	LD	HL,PUMPTAB

	LD	B,SLAVMAx+1	;NUMERO DI ENTRIES DA TRATTARE



PRETAB0:INC	HL

	LD	A,'0'		;IL PRIMO MESSAGGIO INVIATO E' '0'

	LD	(HL),A

	INC	HL

	INC	HL

	DJNZ	PRETAB0

	POP	BC

	JR	NOSETR



NOGENRES:



;	ESEGUE INIZIALIZZAZIONI PRE-SCRITTURA RECORD



	LD	A,7		;CONTEGGIA SEMPRE ALMENO

				;SOH,GRUPPO,INDIRIZZO,STX,NMESS,ETX,BCCH,BCCL

	

	LD	(GBTXCNT),A

	LD	HL,UCOM		;RESETTA PUNTATORI

	LD	(LANWRPO),HL

	LD	A,SOH

	LD	(USOH),A

	LD	A,STX

	LD	(USTX),A

	

	POP	DE		;NUMERO DEL RECORD (DESTINAZIONE MESSAGGIO)

	

	LD	A,E



	OR	A		;NON DEVE ESSERE <0

	JP	M,ERR1

	CP	SLAVMAX		;NON DEVE ESSERE > SLAVMAX

	JP	NC,ERR1

	

	LD	(UDEST),A	;FISSA DESTINAZIONE DEL MESSAGGIO



	POP	HL		;RIPRENDI INDIRIZZO DEL BUFFER

	LD	DE,(LANWRPO)	;BUFFER POINTER



LANW1:	LD	A,(GBTXCNT)	;INCREMENTA NUMERO DEI CARATTERI DA TRASMETTERE

	INC	A

	CP	LANBUFL-8	;VEDI SE BUFFER PIENO

	JP	Z,ERR2		;ERRORE, NON PROCEDERE

	LD	(GBTXCNT),A

	

	LD	A,(HL)		;PRENDI CARATTERE

	LD	(DE),A		;E SALVALO NEL BUFFER LAN



	OR	A		;"0" E' IL TERMINATORE

	JR	Z,WENDREC	;SE LO TROVI FINE SCRITTURA RECORD



	CP	'"'		;" E' IL TERMINATORE

	JR	Z,WENDREC	;SE LO TROVI FINE SCRITTURA RECORD



	CP	CR		;CR E' IL TERMINATORE

	JR	Z,WENDREC	;SE LO TROVI FINE SCRITTURA RECORD



	INC	DE		;ANCORA BYTES DA SCRIVERE - INCREM.PUNTATORI

	INC	HL

	

	JR	LANW1		;E LOOP

	

WENDREC:LD	A,ETX		;INVECE DELLO ZERO SCRIVIAMO ETX

	LD	(DE),A

	

	INC	DE		;PREDISPONIAMO PER BCC

	

	LD	(LANWRPO),DE	;SALVA PUNTATORE DI SCRITTURA GIA'

				;INCREMENTATO - SERVIRA' DOPO PER APPENDERE

				;IL BCC

				

				;AVVIO RICHIESTA TRASMISSIONE

	LD	A,(BUF2)	;ERA ANCORA PENDENTE UNA PRECEDENTE

	OR	A		;RICHIESTA? ERRORE!

	JP	NZ,ERR3

	

	LD	A,(MREADY)	;ERA ANCORA PENDENTE UNA RISPOSTA?

	OR	A

	JP	NZ,ERR4		;ERRORE!

	

	LD	A,WAITING

	LD	(MREADY),A

	

	LD	A,1		;AVVIA RICHIESTA

	LD	(PRE_BUF2),A

	

	XOR	A

	RET



ERR1:	LD	A,1		;NUMERO STAZIONE INESISTENTE

	JR	ERR

ERR2:	LD	A,2		;RECORD TROPPO LUNGO

	JR	ERR

ERR3:	LD	A,3		;WRITE RIPETUTO PRIMA DELL'ARRIVO DELLA RISP

	JR	ERR

ERR4:	LD	A,4		;WRITE RIPETUTO PRIMA DEL READ

	JR	ERR

ERR5:	LD	A,5		;ARRIVATA RISPOSTA SENZA ESSERE IN ATTESA

		

ERR:	DI

	JR	ERR		





;----------------------------------------------------------------------------

;	FUNZIONE N.1 - BLOCK READ

;----------------------------------------------------------------------------

DRV_RD:	LD	A,(MREADY)	;ATTENDI DATI SE NON PRONTI

	CP	READY

	JR	NZ,DRV_RD

	

	EX	DE,HL		;DESTINAZIONE DATI IN DE

	LD	HL,READBUF

	LD	BC,(CNTREAD)

	LDIR

	LD	A,CR		;TERMINATORE

	LD	(DE),A

	XOR	A		;OK PER USCITA DRIVER

	LD	(MREADY),A	;AZZERA FLAG DI MESSAGGIO IN ATTESA

	RET





;----------------------------------------------------------------------------

;	FUNZIONE N.2 - INIZIALIZZAZIONE

;----------------------------------------------------------------------------



INIT:	LD	A,C		;FATTI UNA COPIA DEL TUO NUMERO

	LD	(CHNUM),A	;DI CANALE - SERVIRA' PER CHIEDERE INT

	

;	FASE 1 - SI ACQUISISCE LA STRINGA DI CONFIGURAZIONE E SI PROGRAMMA

;	LA SIO LAN



;	LD	A,1

;	CALL	LEDOUT

	

	LD	A,7		;IL REG E E' GIA' SETTATO DA CHI HA FATTO

				;CALL DRIVER

	LD	D,'S'		;ASPETTATI UNA SIO

	CALL	SYSTEM		;"TRIPLEX" - ANALIZZA LA STRINGA DI CONFIG

				;CHE DEVE ESSERE PRESENTE E RITORNA

				;D=DATA C=CONTROLLO E=VETTORE B=TABELLA

	

;	LD	A,0

;	CALL	LEDOUT

	

	LD	HL,CTLP		;CONTROL PORT

	LD	(HL),C

	INC	HL

	LD	(HL),D		;DATA PORT

	INC	HL

	LD	(HL),E		;VECTOR PORT

				;B=N.TABELLA

				

	LD	A,3		;CHIAMA INSTVEC

	LD	HL,TABVECS	;VETTORI DA INSTALLARE

	LD	E,0		;SIO

	CALL	SYSTEM

	

	PUSH	BC		;VETTORE

	



;	VEDI LA NOTA TECNICA "RESET.DOC" PER INFORMAZIONI RELATIVE ALLE

;	PROCEDURE SEGUITE QUI SOTTO PER IL RESET DELLE PORTE ECC.





	LD	A,(CTLP)

	LD	C,A



	LD	A,18H		;RESETTA CANALE CTLP

	OUT	(C),A



	LD	A,(VECP)

	LD	C,A



	LD	A,1		;PROGRAMMA REGISTRO 1

	OUT	(C),A		;SE ERA GIA' PROGRAMMATO, PACE



	LD	A,14H		;CON SOLO "STATUS AFFECTS VECTOR"

	OUT	(C),A		;E PARITY AFFECTS VECTOR

				;DOPO PROVATO SOSTITUIRE CON UNA LETTURA

				;PRIMA DELLA SCRITTURA

	

	LD	HL,TABSIO

	LD	A,(CTLP)

	LD	C,A

	LD	B,13		;ELEMENTI IN TABELLA



	OTIR



	LD	A,(VECP)

	LD	C,A



	LD	A,2		;SELEZIONA REGISTRO VETTORE

	OUT	(C),A



	POP	DE		;VETTORE - PUSHATO COME BC

	LD	A,E		;PROGRAMMA VETTORE

	OUT	(C),A

	

	IF	SLAN		

				;TEST ONLY

;	LD	A,0FFH		;BIT MODE

;	OUT	(PIOAC),A

	

;	XOR	A

;	OUT     (PIOAC),A

;	CALL	LEDOUT		;SPEGNI TUTTO       

	ENDIF

        

	CALL	SETRX



;	FASE 2 - SI INIZIALIZZANO LE VARIABILI



	LD	HL,ROMSTR	;COPIA IL MESSAGGIO DI DEFAULT (POLLING)

	LD	DE,LBUF		;DA ROM IN RAM

	LD	BC,8

	LDIR



	LD	A,IDLE		;PROCESSO TIMER IN ATTESA

	LD	(TIMSTAT),A

	

	LD	A,ST0

	LD	(RXSTAT),A	;RICEZIONE IN STATO ST0

	

	XOR	A

	LD	(BUF2),A	;TRASMISSIONE DA BUFFER DI DEFAULT

	LD	(GBTXCNT),A	;CONTATORE CARATTERI SCRITTI DA GBASIC

				;NEL PROPRIO BUFFER

	LD	(PRE_BUF2),A

				

	INC	A		;FASE INIZIALE E' STARTUP

	LD	(STARTUP),A

	

	LD	A,REPMAX	;DEFAULT

	LD	(MAXRTY),A



	LD	A,ETX		;METTI I FERMI A FINE BUFFER

	LD	(RBLAST),A

	LD	(UBLAST),A

	

;	PREDISPOSIZIONE INIZIALE TABELLA CON NUM MESSAGGI

	

	LD	A,0FFH		;-1 - I VALORI SARANNO POI INCREMENTATI DA

				;NEXTON E NEXTOFF

	LD	(LASTON),A	;STAZIONE ON LINE DA CUI INIZIARE

	LD	(LASTOFF),A	;STAZIONE OFFLINE DA CUI INIZIARE

	

	LD	HL,PUMPTAB

	

	LD	B,SLAVMAX+1	;MODIFICA 20/10/91



PRETAB:	LD	A,'1'		;PRIMO MESSAGGIO ATTESO E' '1'

	LD	(HL),A

	INC	HL



;	NUMERO DI MESSAGGIO INVIATO AGGIORNATO SOLO AL RESET GENERALE



	INC	HL

	LD	A,ONLINE	;TUTTE LE STAZIONI ONLINE AL PRIMO GIRO

	LD	(HL),A

	INC	HL

	DJNZ	PRETAB

	

	LD	A,ONOFFR	;PREDISPONE RAPPORTO INTERROGAZIONI AD ON/OFF

	LD	(ONCOUNT),A



	XOR	A

	LD	(LANRXCNT),A

	LD	(LANTXCNT),A

	LD	(MREADY),A



	LD	HL,LANRXBU

	LD	(LANRDPO),HL

	LD	(LANRXPO),HL



	LD	HL,UCOM		;AREA TESTO BUFFER UTENTE 

	LD	(LANWRPO),HL



;	FASE 3 - INSTALLA PROCESSO MMTC

	

	LD	A,10		;INSTALLA CHIAMATA PERIODICA PROCESSO MMTC

	LD	HL,MMTC

	CALL	SYSTEM



	XOR	A

	EI

	RET





;----------------------------------------------------------------------------

;	FUNZIONE N.6 - STATUS

;----------------------------------------------------------------------------



STATUS:	LD	A,E		;E = NUMERO STAZIONE

	OR	A		;NON DEVE ESSERE <0

	JP	M,ERR1

	CP	SLAVMAX+1	;NON DEVE ESSERE > SLAVMAX

	JP	NC,ERR1

	

	CALL	QPSTAT

	LD	E,0		;ON LINE

	LD	A,0

	RET	NC

	LD	E,1		;OFF LINE

	RET

	

;--------------------------------------------------------------------







;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



;	PROCESSO TRASMETTITORE (MMTC) - ESEGUE L'INVIO DEI MESSAGGI

;	DI INTERROGAZIONE



;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



MMTC:	PUSH	AF

	PUSH	BC

	PUSH	DE

	PUSH	HL



MMTCA:	LD	A,(PRESC100)	;DECREMENTA PRESCALER

	DEC	A

	LD	(PRESC100),A

	CALL	Z,PROCLAN	;PROCLAN PUO' ALTERARE OGNI REGISTRO

	POP	HL

	POP	DE

	POP	BC

	POP	AF

	RET



;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



;			PROCESSO RICEVITORE (LANRX)



;	SI TRATTA DI UN SOTTOPROCESSO ATTIVATO AD OGNI ARRIVO DI CARATTERE

;	E NON CON L'USO DI UN CLOCK



;	GLI STATI SONO DETERMINATI DA RXSTAT E POSSONO VALERE



ST0	EQU	0	;ATTESA SXT

ST1	EQU	1	;USATO DA GLAN

ST1_2	EQU	2	;USATO DA GLAN

ST1_3	EQU	3	;USATO DA GLAN

ST2	EQU	4	;ATTESA N. DI MESSAGGIO

ST3	EQU	5	;RIEMPIMENTO BUFFER MESSAGGIO

ST4	EQU	6	;ATTESA BCC

ST4_1	EQU	7	;ATTESA BCC

ST5	EQU	8	;MESSAGGIO RICEVUTO





;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



;----------------------------------------------------------------------------

;	LANRX - PROCESSO RICEVITORE

;----------------------------------------------------------------------------



;	IL SUO COMPORTAMENTO E' DIPENDENTE DALLO STATO DELLA

;	MACCHINA (VARIABILE RXSTAT)





LANRX:	EX	AF,AF'		;XCG REGISTERS

	EXX

	

	LD	A,(DTAP)	;NUMERO DI PORTA IN C

	LD	C,A

	

	IN	A,(C)		;LEGGI IL DATO

	AND	7FH		;VIA LA PARITA'

	PUSH	AF		;SERVIRA' DOPO

	

	LD	A,(RXSTAT)	;IN QUALE STATO CI TROVIAMO?



	CP	ST0

	JP	Z,RXST0



	CP	ST1

	JP	Z,RXST1

	CP	ST1_2

	JP	Z,RXST1_2

	CP	ST1_3

	JP	Z,RXST1_3

	

	CP	ST2

	JP	Z,RXST2

	CP	ST3

	JP	Z,RXST3

	CP	ST4

	JP	Z,RXST4

	

	CP	ST4_1

	JP	Z,RXST4_1

	

	CP	ST5

	JP	Z,RXST5

	JP	INT_ERR		;SITUAZIONE NON PREVISTA



;----------------------------------------------------------------------------

;	STATO ST0 - AZZERAMENTO E ATTESA STX, ACK O NAK

;----------------------------------------------------------------------------



RXST0:	CALL	RXRESET		;AZZERA VARIABILI DI RICEZIONE



	POP	AF		;RIPRENDI DATO LETTO

	CP	SOH

	JP	Z,RXSOH		;SE NON E' SOH VEDI SE E' ACK O NAK

	CP	ACK

	JR	Z,RST0A

	CP	NAK

	JP	NZ,RXEXIT	;IGNORA TUTTO SE NON ACK, NAK O STX



RST0A:	CALL	STOTOB		;ACK E NAK VENGONO TRATTATI ALLO STESSO MODO

	LD	A,ST5

	LD	(RXSTAT),A	;SALTA SUBITO A FINE CICLO

	JP	RXST5B

	

RXSOH:				;E' SOH

	CALL	STOTOB		;METTILO NEL BUFFER

	LD	A,ST1

	LD	(RXSTAT),A	;PASSA AD ATTENDERE GROUP ADDRESS 

	JP	RXEXIT



;----------------------------------------------------------------------------

;	STATO ST1 - ATTESA INDIRIZZO ALTO

;----------------------------------------------------------------------------

RXST1:	POP	AF

	CP	SOH

	JP	Z,TOST1		;RESTART?



	CP	'0'		;IN QUESTO STATO E' AMMESSO SOLO

	JP	C,TOST1		;DI RICEVERE UNA CIFRA (AD)

	CP	'@'

	JP	NC,TOST1



	CALL	STOTOB		;REGISTRA GA



RXST1A:	LD	A,ST1_2

	LD	(RXSTAT),A	;PASSA AD ATTENDERE ID

	JP	RXEXIT



;----------------------------------------------------------------------------

;	STATO ST1_2 - ATTESA INDIRIZZO BASSO 

;----------------------------------------------------------------------------

RXST1_2:POP	AF

	CP	SOH

	JP	Z,TOST1		;RESTART?

	

	CP	'0'		;IN QUESTO STATO E' AMMESSO SOLO

	JP	C,TOST1		;DI RICEVERE UNA CIFRA (AD)

	CP	'@'

	JP	NC,TOST1

	CALL	STOTOB		;REGISTRA AD



	LD	A,(RGA)		;PRENDI GROUP ADDRESS

	SUB	'0'		;RENDILO BINARIO

	LD	B,A		;SALVA

	SLA	A		;MOLTIPLICA PER 2

	SLA	A		;MOLTIPLICA PER 4

	SLA	A		;MOLTIPLICA PER 8

	ADD	A,B		;PER 9

	ADD	A,B		;PER 10

	LD	B,A		;SALVA



	LD	A,(RID)		;CIFRA MENO SIGN.

	SUB	'0'		;->BINARIO

	ADD	A,B		;QUESTA E' LA STAZIONE INTERROGATA

	LD	B,A		;SALVA ANCORA

	LD	A,(PDEST)

	CP	B

	JP	NZ,TOST0	;NON E' QUELLA GIUSTA - IGNORA



	LD	A,ST1_3

	LD	(RXSTAT),A	;PASSA AD ATTENDERE NMSG

	JP	RXEXIT



;----------------------------------------------------------------------------

;	STATO ST1_3 - ATTESA NUMERO DI MESSAGGIO

;----------------------------------------------------------------------------



RXST1_3:POP	AF



	CP	SOH		;SE RICEVI ANCORA SOH

	JP	Z,TOST1		;RICOMINCIA DA 1



	CP	'0'		;IN QUESTO STATO E' AMMESSO SOLO

	JR	C,TOST0		;DI RICEVERE UNA CIFRA (N.MESSAGGIO)

	CP	'9'+1

	JR	NC,TOST0

	CALL	STOTOB		;REGISTRA N.DI MESSAGGIO

RXST2A:	LD	A,ST2

	LD	(RXSTAT),A	;PASSA AD ATTENDERE STX

	JP	RXEXIT



;----------------------------------------------------------------------------

;	STATO ST2 - ATTESA STX 

;----------------------------------------------------------------------------



RXST2:	POP	AF		;ATTENDE STX

	CP	SOH

	JP	Z,TOST1

	CP	STX

	JP	NZ,TOST0	;NON STX - TORNA ALLO STATO ST0

	CALL	STOTOB

	LD	A,ST3		;PASSA AD ATTENDERE TESTO MESSAGGIO

	LD	(RXSTAT),A

	JP	RXEXIT



;----------------------------------------------------------------------------

;	SETRXERR - SETTA CONDIZIONI DI ERRORE NELLA RICEZIONE

;----------------------------------------------------------------------------

;SETRXERR:

;	LD	A,(RXERR)

;	OR	BCCERR

;	LD	(RXERR),A

;	JP	RXST2A



;----------------------------------------------------------------------------

;	STATO ST3 - MESSAGGIO VERO E PROPRIO

;----------------------------------------------------------------------------



RXST3:	LD	A,WAITMAX	;REINIZIALIZZA CONTATORE STATI WAIT

	LD	(WAITCNT),A	

	POP	AF



	CP	SOH		;SE RICEVI ANCORA SOH

	JP	Z,TOST1		;RICOMINCIA DA 1

	

	CALL	STOTOB		;QUALUNQUE ALTRA COSA SIA SALVALO

	CP	ETX		;HAI SALVATO ETX?

	JP	NZ,RXEXIT	;ESCI

	LD	A,ST4		;PASSA AD ATTENDERE BCC

	LD	(RXSTAT),A

	JP	RXEXIT



;----------------------------------------------------------------------------

;	TOST0 - PASSA DIRETTAMENTE NELLO STATO ST0

;----------------------------------------------------------------------------

TOST0:	LD	A,ST0		;RIPORTA ALLO STATO 0

	LD	(RXSTAT),A

	JP	RXEXIT



;----------------------------------------------------------------------------

;	TOST1 - PASSA DIRETTAMENTE NELLO STATO ST1

;----------------------------------------------------------------------------

TOST1:	CALL	RXRESET		;AZZERA VARIABILI RICEZIONE

	LD	A,SOH

	CALL	STOTOB

	LD	A,ST1		;RIPORTA ALLO STATO 1

	LD	(RXSTAT),A

	JP	RXEXIT

	

;----------------------------------------------------------------------------

;	STATO ST4 - ATTESA BCCH

;----------------------------------------------------------------------------



RXST4:	POP	AF

	CP	SOH

	JP	Z,TOST1		;RESTART?



	CP	'0'		;IN QUESTO STATO E' AMMESSO SOLO

	JP	C,TOST0		;DI RICEVERE UNA CIFRA 30-3F

	CP	'@'

	JP	NC,TOST0

	LD	(BCCH),A



	LD	A,ST4_1		;PASSA AD ATTENDERE LRCL

	LD	(RXSTAT),A

	JP	RXEXIT

	



; 	NOTA: I DUE BCC NON SONO SALVATI NEL BUFFER TANTO E' INUTILE



;----------------------------------------------------------------------------

;	STATO ST4_1 - ATTESA BCCL

;----------------------------------------------------------------------------

	

RXST4_1:POP	AF		;ATTENDE BCCL

	CP	SOH

	JP	Z,TOST1

	

	CP	'0'		;IN QUESTO STATO E' AMMESSO SOLO

	JP	C,TOST0		;DI RICEVERE UNA CIFRA 30-3F

	CP	'@'

	JP	NC,TOST0



				;ADESSO COMBINA I DUE BCC RICEVUTI

	AND	0FH		;IN UN SOLO BYTE

	LD	B,A

	LD	A,(BCCH)

	AND	0FH

	SLA	A

	SLA	A

	SLA	A

	SLA	A



	OR	B		;ACC CONTIENE BCC COMPLETO

	

	LD	HL,RXBCC

	CP	(HL)		;CONFRONTALO CON QUELLO CALCOLATO

	

	LD	A,ST5

	LD	(RXSTAT),A	;VAI NELLO STATO ST5



	JP	Z,RXST5B	;LA CONDIZIONE E' IL CONFRONTO CON BCC

				;VA DIRETTAMENTE NELLO STATO ST5

				;ANCHE SE NON ARRIVA ALCUN CARATTERE

				;OSSERVARE CHE NON SI SALTA DIRETTAMENTE

				;A ST5 MA SI PASSA PRIMA DA ST5B

	



;	????? PERCHE' NON RIMANDARLO A ST0?



	LD	A,(RXERR)

	OR	BCCERR		;SE ERRORE, PRIMA SETTA FLAG

	LD	(RXERR),A

	JP	RXST5B	

;----------------------------------------------------------------------------

;	STATO ST5 - ATTENDE CHE IL PROCESSO TIMER SIA TORNATO IN WAITING

;----------------------------------------------------------------------------





;	NOTA: IN QUESTO STATO CI ARRIVA SUBITO DOPO AVERE RICEVUTO IL BCC

;	      ANCHE SE NON ARRIVA ALCUN CARATTERE



;	OGNI CARATTERE IN ARRIVO E' IGNORATO FINCHE' LO STATO DI TIMER

;	NON DIVIENE "WAITING"



RXST5:	POP	AF		;WAITING ?! non dovrebbe essere IDLE

RXST5C:	

	LD	A,(TIMSTAT)	;E ATTENDE CHE TIMER LO CAMBI ANCORA

	CP	IDLE		;SENZA FARE NIENTE

;	CP	WAITING		;SENZA FARE NIENTE

	JP	NZ,RXEXIT	



	JP	TOST0		;ABILITA REINIZIO DEL CICLO



;	PASSA DI QUI SOLO ALL'INGRESSO IN ST5 COSI' NON CORRE IL RISCHIO

;	DI MODIFICARE WAITING->RECEIVED IN CASO DI UN CARATTERE SPARSO



RXST5B:	LD	A,RECEIVED

	LD	(TIMSTAT),A	;CAMBIA LO STATO AL PROCESSO TIMER

	JP	RXST5C	



;----------------------------------------------------------------------------

;	RXEXIT - USCITA COMUNE PER TUTTI GLI STATI

;----------------------------------------------------------------------------



RXEXIT:	EXX

	EX	AF,AF'

	EI

	RETI



;----------------------------------------------------------------------------

;	RXRESET	- RESETTA TUTTE LE VARIABILI DI RICEZIONE

;----------------------------------------------------------------------------



RXRESET:LD	A,NOERR		;AZZERA EVENTUALE STATO DI ERRORE

	LD	(RXERR),A

	XOR	A

	LD	(RXBCC),A	;AZZERA BCC PROGRESSIVO PER CONTROLLO



	LD	(LANRXCNT),A	;AZZERA CONTATORE CARATTERI RICEVUTI

	LD	HL,LANRXBU	;RIPORTA PUNTATORE AD INIZIO BUFFER

	LD	(LANRXPO),HL

	RET

	

	

;----------------------------------------------------------------------------

;	ROUTINE STORE TO BUFFER - PONE ACC NEL BUFFER DI RICEZIONE

;----------------------------------------------------------------------------



;	ACC RESTA INTATTO

	

STOTOB:	PUSH	AF

	LD	A,(LANRXCNT)	;INCREMENT BUFFER COUNTER

	INC	A

	LD	(LANRXCNT),A

	CP	LANBUFL+1	;VEDI SE BUFFER PIENO - PUO' AVVENIRE PER

	JP	NC,RXBOER	;ERRORE SLAVE O PER DISTURBO PRESO COME CAR.

	

	POP	AF		;RIPRENDI DATO LETTO



	LD	HL,(LANRXPO)	;GET BUFFER POINTER

	LD	(HL),A		;STORE CHAR

	INC	HL

	LD	(LANRXPO),HL	;STORE BUFFER POINTER

	LD	E,A		;SALVA ACC

	LD	A,(RXBCC)

	XOR	E		;CALCOLA E SALVA NUOVO BCC

	LD	(RXBCC),A

	LD	A,E		;RIPRISTINA ACC

	RET



;	RX BUFFER OVERFLOW ERROR



;	SE SONO ARRIVATI TROPPI CARATTERI IGNORA IL MESSAGGIO CORRENTE

;	(NOTA: STOTOB E' CHIAMATA IN TOST2 MA SUBITO DOPO UN RXRESET QUINDI

;	SENZA PERICOLO DI OVERFLOW - QUESTA USCITA AVVIENE SEMPRE PERTANTO

;	QUANDO E' STATO USATO UN SOLO LIVELLO DI STACK)



RXBOER:	POP	AF		;RIPRISTINA STACK - PUSH AF

	POP	HL		;RIPRISTINA STACK - CALL A STOTOB

	JP	TOST0

	

;----------------------------------------------------------------------------

;	EXT/STATUS CHG - NON DEVE AVVENIRE

;----------------------------------------------------------------------------



LANSTAT:EX	AF,AF'

	EXX	



	LD	A,(CTLP)

	LD	C,A

	LD	A,10H		;RESET EXTERNAL STATUS/INTERRUPT

	OUT	(C),A

	

;	LD	A,1

;	OUT	(C),A

;	LD	A,16H

;	OUT	(C),A

	

	JR	QINT



;----------------------------------------------------------------------------

;	CONDIZIONI SPECIALI - DI SOLITO ERRORE PARITA'

;----------------------------------------------------------------------------



LANSPEC:EX	AF,AF'

	EXX	



	LD	A,(CTLP)	;PRENDI NUMERO DELLA PORTA CONTROLLO

	LD	C,A		;IN C



	LD	A,1		;INDAGA SUL TIPO DI ERRORE

	OUT	(C),A		;(SOLO PER DEBUG) - LEGGI REG1.

	IN	A,(C)

	

	LD	A,30H		;ERROR RESET 

	OUT	(C),A

	

	LD	A,(RXERR)	;SETTA FLAG DI ERRORE PARITA'

	SET	1,A

	LD	(RXERR),A

QINT:	EX	AF,AF'

	EXX

	EI

	RETI



;----------------------------------------------------------------------------

;	LANTX - TRASMISSIONE CARATTERE VERSO LAN - NON SERVE UN INTERR

;----------------------------------------------------------------------------



;	MA VIENE CHIAMATA DAL PROCESSO AD 1 MSEC

;	NEI MOMENTI IN CUI NON E' ATTIVA LA TRASMISSIONE 

;	PUO' ESSERE CHIAMATA ANCHE DAI PROGRAMMI



;LANTX:	PUSH	BC

;	PUSH    AF

;	LD	A,(CTLP)	;NUMERO DI PORTA IN C

;	LD	C,A

	

;SIOT1:	IN	A,(C)

;	BIT	2,A		;TEST TX EMPTY

;	JR	Z,SIOT1

;	LD	A,(DTAP)	;NUMERO PORTA DATI IN C

;	LD	C,A

;       POP     AF

;	OUT	(C),A

	

;	POP	BC

;	RET



;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



;	PROCESSO LAN (PROCLAN) - ESEGUE LE INTERROGAZIONI AGLI SLAVES



;	VIENE CHIAMATO A TEMPO



;	GLI STATI POSSIBILI SONO DETRMINATI DALLA VARIABILE TIMSTAT

;	E SONO



IDLE	EQU	0

WAITING	EQU	1

RECEIVED EQU	2

READY	EQU	3



;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



PROCLAN:LD	A,(TIMSTAT)

	

	CP	IDLE

	JP	Z,TIMIDLE



	CP	WAITING

	JP	Z,TIMWTNG



	CP	RECEIVED

	JP	Z,TIMRCVD



	JP	TIMEXIT		;ERA INT_ERR 20/10/91





;	SEGUONO GLI STATI DEL PROCESSO TIMER;



;----------------------------------------------------------------------------

;	STATO IDLE - VIENE AVVIATA LA TRASMISSIONE

;----------------------------------------------------------------------------



;	SE NON CI SONO TRASMISSIONI IN CORSO, UNA NUOVA TRASMISSIONE

;	VIENE AVVIATA USANDO IL BUFFER ATTIVO, OSSIA IL

;	BUFFER DI POLLING (SE IL BASIC NON HA RICHIESTO DI TRASMETTERE)

;	O DA QUELLO DEL BASIC (SE QUESTO HA RICHIESTO DI TRASMETTERE)



TIMIDLE:LD	A,(STARTUP)	;NON INVIARE INTERROGAZIONI FINCHE'

	OR	A		;SEI IN STARTUP

	JP	NZ,TIMEXIT





;	----- SI VERIFICA CHE NON CI SIANO TRASMISSIONI IN CORSO



	LD	A,(LANTXCNT)	;VERIFICA CHE NON CI SIA ANCORA

	OR	A		;DA TRASMETTERE QUALCOSA

	JP	NZ,TIMEXIT	;IN QUESTO CASO RITORNA SUBITO

	

;	----- NESSUNA TRASMISSIONE E' IN CORSO - SE NE PUO' INIZIARE UNA NUOVA

	

				;LE SEGUENTI INIZIALIZZAZIONI SONO POSTE QUI

				;PER NON ESEGUIRLE NELLE RITRASMISSIONI

				

	LD	A,(MAXRTY)	;INIZIALIZZA CONTATORI

	LD	(REPCNT),A	;RITRASMISSIONI

	LD	(NAKCNT),A	;E NAK INVIABILI

	

		

;	----- DECIDE QUALE BUFFER E' QUELLO ATTIVO



	LD	A,(PRE_BUF2)	;VEDE SE BASIC HA RICHIESTO DI TRASMETTERE

	OR	A

	JP	NZ,STARTGB	;SALTA PER TRASMETTERE MSG RICH. DA GBASIC

	

;----------------------------------------------------------------------------

;	ANCORA IDLE: AVVIA PROCESSO TRASMISSIONE MESSAGGIO DI DEFAULT

;----------------------------------------------------------------------------



;	----- PREDISPONE BUFFER



;	----- DECISIONE STAZIONE DA INDIRIZZARE



;	PER PRIMA COSA SI VEDE SE DEVE ESSERE INTERROGATA UNA STAZIONE

;	ONLINE O UNA STAZIONE OFFLINE



	LD	A,(ONCOUNT)	;ONCOUNT STABILISCE IL RAPPORTO TRA LE

	DEC	A		;INTERROGAZIONI ALLE ON ED ALLE OFFLINE

	OR	A		;QUANDO ONCOUNT=0 SI INTERROGA UNA OFF

	LD	(ONCOUNT),A	;SALVA, POI SI VEDRA'

	JP	NZ,POLLON	;E' L'ORA DI INTERROGARE LA PROSSIMA OFF?

				;SALTA SE NO

	LD	A,ONOFFR	;SI, RIPRISTINA CONTATORE

	LD	(ONCOUNT),A



POLLOFF:CALL	NEXTOFF		;DETERMINA IL NUMERO DELLA PROSSIMA OFF

	JP	C,POLLON	;SE NON CI SONO STAZIONI IN OFF PROVA CON ON

	LD	(LASTOFF),A	;AGGIORNA PUNTATORE

	LD	(PDEST),A	;PDEST RICORDA QUALE E' LA STAZ. INTERROGATA



	PUSH	HL

	LD	HL,LDESTH	;INSERISCI INDIRIZZO DEST NEL BUFFER

	CALL	SETDEST

	POP	HL



	LD	A,OFFLINE	;FLAG USATO PER RICONOSCERE I "RISVEGLI"

	LD	(WAKEUP),A	;INDICA CHE L'INTERROGAZIONE E' STATA FATTA

	JP	SETNMSG		;AD UNA STAZIONE OFF

	

POLLON:	CALL	NEXTON		;PROCURATI IL NUMERO DELLA PROSSIMA STAZIONE ON

	JP	C,POLLOFF

	LD	(LASTON),A

	LD	(PDEST),A	;PDEST RICORDA QUALE E' LA STAZ. INTERROGATA

	

	PUSH	HL

	LD	HL,LDESTH	;INSERISCI INDIRIZZO DEST NEL BUFFER

	CALL	SETDEST

	POP	HL

	

	LD	A,ONLINE

	LD	(WAKEUP),A

	

;	INSERISCI NUMERO DI MESSAGGIO - ESSO TUTTAVIA SARA' INCREMENTATO

;	SOLO DOPO AVER RICEVUTO LA RISPOSTA



SETNMSG:			;NXTON O NXTOFF HANNO LASCIATO HL A PUNTARE

				;ALLA LOCAZIONE DELLO STATUS DELLA TABELLA STAZIONI



	DEC	HL		;PORTALO SU NUMERO DEL MESSAGGIO DA SPEDIRE

	LD	A,(HL)

	LD	(LNMSG),A	;INSERISCILO NEL BUFFER

	

	DEC	HL		;PORTA HL SU NUMERO MESSAGGIO DA RICEVERE

	LD	A,(HL)

	LD	(EXPNMSG),A	;SALVALO PER USO FUTURO



;	----- MESSAGGIO PREDISPOSTO - VALUTA LUNGHEZZA, AZZERA PUNTATORI

;				      E CALCOLA BCC



REIDLE:	;ENTRY POINT PER RITRASMISSIONE STESSO MESSAGGIO SENZA INCREMENTO

;------	 DEL NUMERO DI MESSAGGIO

	

	LD	A,WAITMAX	;REINIZIALIZZA CONTATORE STATI WAIT

	LD	(WAITCNT),A	



	LD	A,(BUF2)	;VEDI QUALE BUFFER DEVI SPEDIRE

	OR	A

	JP	NZ,STGB2

	

;	INVIO INTERROGAZIONE DI POLLING



	LD	A,(.LOW.(LBCC-LSOH))+2	;LUNGHEZZA BUFFER DA TRASMETTERE

	LD	(LANTXCNT),A

	LD	HL,LBUF		;INIZIALIZZA PUNTATORI ALL'AREA 

				;DEL MESSAGGIO DI DEFAULT

	LD	(LANTXPO),HL

	CALL	BCC		;BCC DA SOH INCLUSO

	CALL	ADD_BCC		;APPENDILO AL BUFFER

	

;	----- BUFFER COMPLETO - CAMBIA STATO

	

BCOMPL:	CALL	STARTX		;AVVIA TRASMISSIONE



;	TENTATIVO DI MODIFICA - RIPRISTINATO PERCHE' NON VA:



;	MODIFICA 20/10/91 - INVECE DI PASSARE SUBITO IN WAITING SI PASSA

;	IN UNO STATO DI SOSPENSIONE IN CUI NON SI FA NIENTE - SARA' LA

;	FINE DELLA TRASMISSIONE A ENTRARE IN WAITING



	LD	A,WAITING	;PASSA IN ATTESA DI RISPOSTA

	LD	(TIMSTAT),A

	

	LD	A,ST0		;ABILITA RICEZIONE

	LD	(RXSTAT),A



	JP	TIMEXIT





;----------------------------------------------------------------------------

;	ANCORA IDLE: AVVIA TRASMISSIONE MESSAGGIO BASIC

;----------------------------------------------------------------------------



;	----- PREDISPOSIZIONE BUFFER GBASIC ED AVVIO TRASMISSIONE

	

STARTGB:LD      A,(UDEST)       ;NUMERO DELLA STAZIONE RICHIESTA

	LD	(PDEST),A	;AGGIORNA PDEST

        CALL    QPSTAT          ;ESAMINA STATO



;	MODIFICA SPERIMENTALE 3/9/90 - INVIO MESSAGGIO GBASIC

;	A STAZIONE OFFLINE RITORNA DIRETTAMENTE UN TIMEOUT - IN QUESTO MODO

;	SIAMO SICURI CHE ALLO SLAVE NON SIA RIMASTO UN MESSAGGIO "IN CANNA"



	JP	NC,PRETMO2



;	LD      A,OFFLINE

;	JP      NC,STGB1



	LD      A,ONLINE

STGB1:  LD      (WAKEUP),A      ;MEMORIZZA STATO ATTUALE PER DETERMINARE

                                ;EVENTUALE RISVEGLIO

				;QPSTAT HA LASCIATO HL A PUNTARE

				;ALLA LOCAZIONE DELLO STATUS DELLA TABELLA STAZIONI

	DEC	HL		;PORTALO SU NUMERO DEL MESSAGGIO DA SPEDIRE

	LD	A,(HL)

	LD	(UNMSG),A	;INSERISCILO NEL BUFFER

	

	DEC	HL		;ADESSO PRENDI NUMERO MESSAGGIO DA RICEVERE

	LD	A,(HL)

	LD	(EXPNMSG),A	;SALVALO PER USO FUTURO

	

	LD	A,(PDEST)

	LD	HL,UDESTH	;INSERISCI INDIRIZZO DEST NEL BUFFER

	CALL	SETDEST

	

STGB2: 	LD	A,(GBTXCNT)	;NUMERO DI CARATTERI DA TRASMETTERE

	LD	(LANTXCNT),A

	LD	HL,UBUF		;INDIRIZZO BUFFER DA TRASMETTERE

	LD	(LANTXPO),HL



	CALL	BCC		;CALCOLA BCC DA SOH INCLUSO

	CALL	ADD_BCC		;E APPENDILO NEL BUFFER

	

	XOR	A

	LD	(PRE_BUF2),A	;IMPEDISCE CHE RISPOSTE AUTO

	INC	A		;SIANO PRESE COME RISPOSTE BASIC

	LD	(BUF2),A

	JP	BCOMPL



;	VEDI NOTE SOPRA



PRETMO2:XOR	A

	LD	(PRE_BUF2),A	

	INC	A

	LD	(BUF2),A

	CALL	SETTMO

	JP	EXPCTD

	

;----------------------------------------------------------------------------

;	STATO WAITING - SI ATTENDE IL RITORNO DI UNA RISPOSTA

;----------------------------------------------------------------------------



;	SALVO TIMEOUT SI ESCE DA QUESTO STATO PER OPERA DEL PROCESSO

;	RICEVITORE CHE MODIFICA TIMSTAT



TIMWTNG:LD	A,(WAITCNT)	;GUARDA DI NON RESTARE TROPPO A LUNGO

	DEC	A		;IN QUESTO STATO

	LD	(WAITCNT),A

	

	JP	NZ,TIMEXIT	;SE CI SEI STATO TROPPO RIPETI



REPEAT:	LD	A,(REPCNT)

	DEC	A

	LD	(REPCNT),A

	JP	Z,TIMOUT

	

	JP	REIDLE





;----------------------------------------------------------------------------

;	STATO RECEIVED - PROCESSO DELLE RISPOSTE RICEVUTE

;----------------------------------------------------------------------------



TIMRCVD:LD	A,(RXERR)	;CI SONO STATI ERRORI?

	OR	A

	JP	Z,NOERROR	;NO, SALTA



SNDNAK:	LD	A,(NAKCNT)	;ERRORE DI RICEZIONE -SPEDISCI NAK

	DEC	A		;MA NON PIU' DI MAXRTY VOLTE

	LD	(NAKCNT),A

	JP	Z,TIMOUT



	LD	HL,NAKROM

	CALL	SINGLE

		

;	LD	A,NAK

;	CALL	LANTX		;INVIO NAK

	

	LD	A,WAITING	;DI NUOVO IN ATTESA DI RISPOSTA

	LD	(TIMSTAT),A



	LD	A,(MAXRTY)

	LD	(REPCNT),A

	

	LD	A,ST0

	LD	(RXSTAT),A



	JP	TIMEXIT

	

NOERROR:LD	HL,LANRXBU	;PUNTA AL BUFFER DEI CARATTERI RICEVUTI

	LD	A,(HL)		;PRIMO CARATTERE E' INIZIO MESSAGGIO?



	CP	SOH		;RICEVUTO MESSAGGIO STANDARD

	JP	Z,STDMSG	;PROCESSALO

	

	CP	ACK		;RICEVUTO ACK

	JP	Z,RETACK	;RITORNA '$'



	CP	NAK		;RICEVUTO NAK

	JP	Z,REPEAT	;RIPETI MESSAGGIO



	JR	SNDNAK		;ALTRO CARATTERE - INVIA NAK

	

	

;----------------------------------------------------------------------------

;	RICEVUTO ACK

;----------------------------------------------------------------------------



RETACK:	

;	CALL	INCNMSG		;E' ARRIVATO UN ACK

				;ADESSO PUOI INCREMENTARE NMSG



	LD	A,(BUF2)	;EVITA IL "BUG 8/90" (VEDI DOC.)

	OR	A		;LIMITATAMENTE AD ACK

	JP	Z,TOIDLE	;GLI ACK RICEVUTI SENZA INTERROGAZIONE

				;GBASIC SONO IGNORATI

	

	LD	A,'$'

	LD	(RKEYCH),A

	

	PUSH	HL

	LD	A,(PDEST)	;COPIA IL NUMERO DELLA STAZIONE ATTESA

	LD	HL,RFROM

	CALL	SETDEST

	POP	HL

	

	LD	A,8		;DOPO SI SOTTRAE 6

	LD	(LANRXCNT),A

	

	JP	EXPCTD



;----------------------------------------------------------------------------

;	RICEVUTO MESSAGGIO STANDARD

;----------------------------------------------------------------------------

	

STDMSG:	LD	A,(RKEYCH)	;RISPOSTA "P" PROIBITA 

	CP	'P'		;("BUG 8/90" V.DOC)

	JP	Z,TOIDLE	;MESSAGGIO IGNORATO

	

;	MODIFICA DI PROVA - LA RISPOSTA ACK 

;	E L'INCREMENTO DI NMSG VENGONO POSPOSTI A DOPO

;	L'EFFETTIVA ACQUISIZIONE DEL MESSAGGIO



;	LE STRADE POSSIBILI, DA QUI SONO:

;		- EXPCTD

;		- UNEXPCT

;		- ACKANDRDY



	LD	A,(WAKEUP)	;GUARDA SE SI TRATTA DI UN RISVEGLIO

	CP	ONLINE		

	JP	Z,ANAMSG	;NO, SALTA SENZA AGGIORNARE TABELLA

	

	LD	A,(PDEST)	;RISVEGLIO, AGGIORNARE TABELLA

	CALL	TLKUP

	INC	HL

	INC	HL

	LD	A,ONLINE

	LD	(HL),A

	JP	UNEXPCT		;E RICHIEDERE INTERRUPT AL GBASIC

;

;	QUI SI ANALIZZA IL MESSAGGIO RICEVUTO

;	

ANAMSG:	LD	A,(BUF2)	;DOMANDE FATTE DA GBASIC DEVONO COMUNQUE

	OR	A

	JP	NZ,EXPCTD	;AVERE UNA RISPOSTA

	

	LD	A,(RKEYCH)	;SE L'INTERROGAZIONE NON E' BASIC VUOL

				;DIRE CHE E' AUTO - SE LA RISPOSTA

	CP	'C'		;E' "NULLA DA DIRE"

	JP	Z,ACKANDRDY	;PROSEGUI PER LA STRADA VELOCE



;----------------------------------------------------------------------------

;	MESSAGGIO RICHIESTO CON POLLING - SI FA INTERRUPT AL BASIC

;----------------------------------------------------------------------------



UNEXPCT:CALL	PREDELV		;OPERAZIONI PRE-CONSEGNA

	

	LD	A,13		;SEGMENTO CRITICO

	CALL	SYSTEM



	XOR	A

	CALL	SYSTEM		;CALL PSH_USTK

	

	LD	A,(CHNUM)

	LD	E,A

	LD	A,2

	CALL	SYSTEM		;CALL ASKMYLEV



	LD	E,A		;INT LEV

	LD	A,1

	CALL	SYSTEM		;CALL HIRQ - RICHIEDI INTERRUPT A BASIC



	LD	A,14		;FINE SEGMENTO CRITICO

	CALL	SYSTEM



	LD	A,(RKEYCH)	;NIENTE ACK SE TIMEOUT

	CP	'*'

	JR	Z,TOIDLE



	CALL	INCNMSG		;E' ARRIVATA UNA RISPOSTA COMPLETA

				;ADESSO PUOI INCREMENTARE FUTURO NMSG



	LD	HL,ACKROM	;SPEDISCI ACK

	CALL	SINGLE



TOIDLE:	LD	A,IDLE	

	LD	(TIMSTAT),A	

	JP	TIMEXIT	





;----------------------------------------------------------------------------

;	MESSAGGIO RICHIESTO CON WRITE - SI PREPARA PER READ

;----------------------------------------------------------------------------



EXPCTD:	CALL	PREDELV		;OPERAZIONI PRE-CONSEGNA MESSAGGIO



	LD	A,(MREADY)	;DEVE ESSERE IN ATTESA

	CP	WAITING

	JP	NZ,ERR5



	LD	(CNTREAD),BC	;SALVA TOTALE CARATTERI

	

	LD	DE,READBUF

	LDIR			;COPIA NEL BUFFER DI TRANSITO

	

	LD	A,READY

	LD	(MREADY),A	;MESSAGGIO PRONTO

	

	XOR	A		;RICOMINCIA CON AUTO

	LD	(BUF2),A

	

	LD	A,(RKEYCH)	;NIENTE INCREMENTO SE TIMEOUT

	CP	'*'

	JR	Z,TOIDLE

	

	CALL	INCNMSG		;E' ARRIVATA UNA RISPOSTA COMPLETA

				;ADESSO PUOI INCREMENTARE FUTURO NMSG



;	LD	HL,ACKROM	;SPEDISCI ACK

;	CALL	SINGLE



	JR	TOIDLE



;----------------------------------------------------------------------------

;	MESSAGGIO RICHIESTO E NON PERVENUTO - TIMEOUT

;----------------------------------------------------------------------------



TIMOUT:	LD	A,(WAKEUP)	;VEDI SE AGGIORNARE TABELLA

	CP	OFFLINE		;SE ERA GIA' OFFLINE NON SI DEVE

	JP	Z,SHORTRDY	;FARE NIENTE - SALTA PER PROCEDURA VELOCE



UPDATAB:LD	A,(PDEST)	;AGGIORNAMENTO EFFETTIVO DELLA TABELLA

	CALL	TLKUP

	INC	HL

	INC	HL

	LD	A,OFFLINE

	LD	(HL),A

	

	CALL	SETTMO

	

	LD	A,(BUF2)	;SE IL TIMEOUT E' CONSEGUENTE ALL'INVIO

	OR	A		;DI UN MESSAGGIO DI SELECTING

	JP	NZ,EXPCTD	;SI DEVE AVERE UNA RISPOSTA

	

	JP	UNEXPCT		;SE TIMEOUT IMPREVISTO RICHIEDI INT GBASIC

	

;----------------------------------------------------------------------------

;	USCITA DA TUTTI GLI STATI DI TIMER

;----------------------------------------------------------------------------



TIMEXIT:LD	A,PRESC_K	;RIPRISTINA VALORE INIZIALE PRESCALER

	LD	(PRESC100),A

	RET

	

;	ENTRANDO QUI SI VA DIRETTI IN IDLE SENZA ATTENDERE 

;	PROSSIMO PROCLAN



ACKANDRDY:

	CALL	INCNMSG		;E' ARRIVATA UNA RISPOSTA COMPLETA

				;ADESSO PUOI INCREMENTARE FUTURO NMSG



	LD	HL,ACKROM	;DA QUI SPEDISCI ANCHE ACK

	CALL	SINGLE



SHORTRDY:

	LD	A,IDLE

	LD	(TIMSTAT),A



	LD	A,PRESC_K	;RIPRISTINA COSTANTE DI TEMPO

	LD	(PRESC100),A



	JP	TIMIDLE



;----------------------------------------------------------------------------

;	STATI DI ERRORE DA MEGLIO DEFINIRE

;----------------------------------------------------------------------------



INT_ERR:			;INTERNAL ERROR

	JR	INT_ERR





;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



;			     ROUTINES VARIE



;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



;----------------------------------------------------------------------------

;	PREDELV - OPERAZIONI PRE-DELIVERY

;----------------------------------------------------------------------------



PREDELV:LD	HL,RKEYCH	;PUNTA AL PRIMO CARATTERE UTILE DEL BUF RX



	LD	B,0

	LD	A,(LANRXCNT)



	SUB	6		;NON CONTARE SOH-GA-ID-STX-NMSG-ETX 

				;(BCC GIA' NON CONTATI)

	

	LD	C,A		;BC CONTIENE # DI CARATTERI

	RET



;----------------------------------------------------------------------------

;	SETTMO - INSERISCE NEL BUFFER MESSAGGIO DI TIMEOUT

;----------------------------------------------------------------------------



SETTMO:	LD	A,'*'		;RITORNA MESSAGGIO DI TIMEOUT

	LD	(RKEYCH),A



	LD	A,9		;DOPO SI SOTTRAE 6

	LD	(LANRXCNT),A

	

	PUSH	HL

	LD	A,(PDEST)	;COPIA IL NUMERO DELLA STAZIONE ATTESA

	LD	HL,RFROM

	CALL	SETDEST

	POP	HL

	RET

	

;----------------------------------------------------------------------------

;	INCNMSG - INCREMENTA IN TABELLA I NUMERI DI MESSAGGIO

;----------------------------------------------------------------------------



;	IL PARAMETRO IN INGRESSO E' PDEST



INCNMSG:LD	A,(PDEST)

	CALL	TLKUP		;HL SU NUMERO DI MESSAGGIO ASPETTATO

	

	LD	A,(HL)

	CALL	INCASC		;INCREMENTALO

	LD	(HL),A		;E AGGIORNA TABELLA

	

	INC	HL		;HL SU NUMERO DI MESSAGGIO DA TRASMETTERE



	LD	A,(HL)

	CALL	INCASC		;INCREMENTA NUMERO MESSAGGIO DA SPEDIRE

	LD	(HL),A		;ED AGGIORNA TABELLA

	

	RET





;----------------------------------------------------------------------------

;	BCC - CALCOLA BCC DA (HL) AD ETX

;----------------------------------------------------------------------------

;	CALCOLA IL BCC DEL BUFFER AD HL E LO LASCIA IN ACC

;	VENGONO TRATTATI I BYTES DA (HL) A ETX COMPRESO

;	ETX DEVE ESSERE NECESSARIAMENTE PRESENTE



BCC:	LD	A,(HL)		;PRIMO DATO

	

BCC2:	INC	HL

	XOR	(HL)		;CALCOLA SOMMA LOGICA

	PUSH	AF		;SALVA RISULTATO PROGRESSIVO

	LD	A,(HL)

	CP	ETX		;FINE DEL BUFFER?

	JP	Z,BCC1

	POP	AF

	JP	BCC2



BCC1:	POP	AF

	RET





;----------------------------------------------------------------------------

;	INCASC - INCREMENTA ACCUMULATORE IN ASCII - SEQUENZA: 1234567891234567891..

;----------------------------------------------------------------------------



INCASC:	INC	A

	CP	'9'+1

	RET	NZ

	LD	A,'1'

	RET



;----------------------------------------------------------------------------

;	INCDEST - INCREMENTA N.DEST CON RICIRCOLO  

;----------------------------------------------------------------------------



;	SEQUENZA 0,1,..SLAVMAX,0..SLAVMAX



INCDEST:INC	A

	CP	SLAVMAX+1

	RET	NZ

	XOR	A		;RICOMINCIA DA 0

	RET



;----------------------------------------------------------------------------

;	TLKUP - POSIZIONA HL SULLA TABELLA STAZIONI PER UNA CERTA STAZIONE

;----------------------------------------------------------------------------

;	TABLE LOOK UP - DATO UN NUMERO DI STAZIONE IN ACC RITORNA IN HL

;	L'INDIRIZZO DEL PRIMO ELEMENTO DELLA TABELLA STAZIONI - DE DISTRUTTO



TLKUP:	LD	E,A

	LD	D,0

	LD	HL,PUMPTAB

	ADD	HL,DE

	ADD	HL,DE

	ADD	HL,DE

	RET



;----------------------------------------------------------------------------

;	NEXTOFF - RITORNA IL NUMERO DELLA PROSSIMA STAZIONE IN OFF

;----------------------------------------------------------------------------



;	IL NUMERO DELLA STAZIONE E' RITORNATO IN ACC

;	SE NESSUNA STAZIONE E' IN OFF RITORNA CARRY SETTATO



NEXTOFF:LD	B,SLAVMAX+1

	LD	A,(LASTOFF)

NXTOF1:	CALL	INCDEST		;GUARDA LA STAZIONE SEGUENTE ALL'ULTIMA INDIRIZZ.

	CALL	QPSTAT

	RET	NC

	DJNZ	NXTOF1

	SCF

	RET			;NESSUNA STAZIONE E' OFF - SETTA CARRY



;----------------------------------------------------------------------------

;	NEXTON - RITORNA IL NUMERO DELLA PROSSIMA STAZIONE IN ON

;----------------------------------------------------------------------------



;	IL NUMERO DELLA STAZIONE E' RITORNATO IN ACC

;	SE NESSUNA STAZIONE E' IN ON RITORNA CARRY SETTATO



NEXTON:	LD	B,SLAVMAX+1

	LD	A,(LASTON)

NXTON1:	CALL	INCDEST		;GUARDA LA STAZIONE SEGUENTE ALL'ULTIMA INDIRIZZ.

	CALL	QPSTAT		;SE TORNA CON FLAG SETTATO VUOL DIRE CHE E'

				;ONLINE

	CCF			;COMPLEMENTA FLAG

	RET	NC		;OSSIA RITORNA CON FLAG CY RESETTATO SE

				;LA STAZIONE E' IN ON 

	DJNZ	NXTON1

	SCF

	RET			;NESSUNA STAZIONE E' ON - SETTA CARRY





;----------------------------------------------------------------------------

;	QPSTAT - QUERY PUMP STATUS 

;----------------------------------------------------------------------------

;	DATO IN ACC UN NUMERO VALIDO ASCII DI STAZIONE RITORNA ACC INVAR. E 

;		       CARRY SET   SE STAZIONE ON LINE

;		       CARRY RESET SE STAZIONE OFFLINE

;

QPSTAT:	PUSH	AF

	CALL	TLKUP

	INC	HL

	INC	HL

	LD	A,(HL)		;PRENDI STATO STAZIONE

	OR	A

	JP	Z,ISOFF

	POP	AF

	SCF			;SET CARRY

	RET

ISOFF:	POP	AF

	OR	A		;CLEAR CARRY

	RET



;----------------------------------------------------------------------------

;	STARTX	- AVVIA TRASMISSIONE CARATTERE VERSO SLAN 

;----------------------------------------------------------------------------



;	VIENE CHIAMATA DIRETTAMENTE - NON SERVE INTERRUPT



STARTX:	CALL	PRETX



	LD	A,(LANTXCNT)	;CONTEGGIA IL CARATTERE CHE STIAMO PER TRASM. 

	OR	A

	RET	Z		;BUFFER VUOTO - IGNORA RICHIESTA

	DEC	A	

	LD	(LANTXCNT),A



	LD	HL,(LANTXPO)

	LD	A,(HL)		;PRENDI CARATTERE

	INC	HL

	LD	(LANTXPO),HL	;AGGIORNA PUNTATORE



	LD	HL,DTAP

	LD	C,(HL)

	OUT	(C),A

	

	RET

	

;----------------------------------------------------------------------------

;	ILANTX - TRASMISSIONE CARATTERE VERSO SLAN 

;----------------------------------------------------------------------------



;	SERVE INTERRUPT





ILANTX:	EX	AF,AF'

	EXX

	

	LD	A,(LANTXCNT)	;CONTEGGIA IL CARATTERE CHE STIAMO PER TRASM. 

	OR	A

	JR	Z,NOTX		;BUFFER VUOTO - BASTA CON GLI INTERRUPTS

	DEC	A	

	LD	(LANTXCNT),A

	

	LD	HL,(LANTXPO)

	LD	A,(HL)		;PRENDI CARATTERE

	INC	HL

	LD	(LANTXPO),HL	;AGGIORNA PUNTATORE



	LD	HL,DTAP

	LD	C,(HL)

	OUT	(C),A



	JP	RXEXIT		;ESCE ALLO STESSO MODO DI RX

	

;	------ FINE BUFFER - ARRESTA INTERRUPTS



NOTX:	LD	A,28H		;NON CI SONO PIU' CARATTERI - FERMATI

	LD	HL,CTLP

	LD	C,(HL)

	OUT	(C),A		;(RESET INTERRUPT PENDING)

	

	CALL	POSTTX

	

	JP	QINT



;----------------------------------------------------------------------------

;	SINGLE - AVVIA LA TRASMISSIONE DI UN SOLO CARATTERE

;----------------------------------------------------------------------------

	

;	CARATTERE DA INVIARE DEVE ESSERE PUNTATO DA HL



SINGLE:	LD	(LANTXPO),HL

	LD	A,1

	LD	(LANTXCNT),A

	CALL	STARTX

	RET





;----------------------------------------------------------------------------

;	PRETX - OPERAZIONI DI INIZIO TRASMISSIONE

;----------------------------------------------------------------------------



;	DISABILITA LA RICEZIONE (PER NON RICEVERE SE STESSO)

;	E PREDISPONE BUFFERS VERSO ESTERNO PER TRASMISSIONE





PRETX:	LD	HL,CTLP

	LD	C,(HL)



	CALL	SETTX

	

PRETX1:	LD	A,3		;SIO RX DISABLE

	OUT	(C),A

	LD	A,40H

	OUT	(C),A



	RET



;----------------------------------------------------------------------------

;	POSTTX - OPERAZIONI DI FINE TRASMISSIONE

;----------------------------------------------------------------------------



;	ATTENDE ALL SENT - RIGIRA BUFFER - RIABILITA RICEZIONE



POSTTX:	LD	HL,CTLP

	LD	C,(HL)



	LD	A,1		;ATTENDI "ALL SENT"

	OUT	(C),A		;PRIMA DI RIABILITARE BUFFERS E RICEZIONE

	IN	A,(C)

	BIT	0,A

	JR	Z,POSTTX

	

	CALL	SETRX



POSTTX1:LD	A,3		;SIO RX ENABLE

	OUT	(C),A

	LD	A,41H

	OUT	(C),A



	RET





;----------------------------------------------------------------------------

;	SETRX - GIRA IL BUFFER IN RICEZIONE

;----------------------------------------------------------------------------

SETRX:	LD	A,(CTLP)

	LD	C,A

	LD	A,5		;REG.5

	OUT	(C),A

	

	LD	A,02AH		;RTS VALIDO (OSSIA /RXEN E TXEN BASSI)

	OUT	(C),A



	RET



;----------------------------------------------------------------------------

;	SETTX - GIRA IL BUFFER IN TRASMISSIONE

;----------------------------------------------------------------------------

SETTX:	LD	A,(CTLP)

	LD	C,A

	LD	A,5		;REG.5

	OUT	(C),A

	

	LD	A,028H		;RTS INVALIDO (OSSIA /RXEN E TXEN ALTI)

	OUT	(C),A



	RET



;----------------------------------------------------------------------------

;	ADD_BCC - AGGIUNGI AL BUFFER I DUE BCC ASCII (BCC DEVE ESSERE IN ACC)

;----------------------------------------------------------------------------

;	IL BUFFER DEVE ESSERE PUNTATO DA HL

;	BBCH -> HL+1;	BCCL -> HL + 2;	 HL = HL + 2;



;	A,B MODIFICATI



ADD_BCC:LD	B,A

	AND	0F0H		;PARTE ALTA BCC

	SRL	A

	SRL	A

	SRL	A

	SRL	A

	ADD	A,30H

	INC	HL		;HL PUNTAVA A ETX

	LD	(HL),A

	LD	A,B

	AND	0FH		;PARTE BASSA BCC

	ADD	A,30H

	INC	HL

	LD	(HL),A		;ABBIAMO AGGIUNTO I DUE BCC NEL BUFFER

	RET



;----------------------------------------------------------------------------

;	SETDEST - CONVERTE ACC IN UNA STRINGA DI DUE CIFRE AD (HL) E (HL+1)

;----------------------------------------------------------------------------

;	PRIMA LA PIU' SIGNIFICATIVA - ACC DEVE ESSERE TRA 0 E 31



SETDEST:CP	10		;E' MINORE DI 10?

	JR	NC,SET10	;SALTA

	PUSH	AF

	LD	A,'0'		;METTI 0 NELLA PARTE ALTA

	LD	(HL),A

	POP	AF

	INC	HL

	ADD	A,'0'

	LD	(HL),A

	RET

	

SET10:	PUSH	DE

	EX	DE,HL		;DE = DESTINAZIONE

	LD	H,0

	LD	L,A

	LD	A,18		;CALL CNS

	CALL	SYSTEM

	POP	DE

	RET

	

;	IF	SLAN

;----------------------------------------------------------------------------

;	LEDOUT - EMETTE UN DATO SUI LED DA ACC

;----------------------------------------------------------------------------

;LEDOUT:	AND	0FH		;SOLO 4 LED MENO SIGN.

;	PUSH	BC

;	LD	B,A		;SALVA DATO PER LED



;	IN	A,(PIOAD)	;PER VEDERE MEGLIO ALLO OSCILLOSCOPIO

;	XOR	B		;COMPLEMENTA BIT RICHIESTI

;	OUT	(PIOAD),A



;	IN	A,(PIOAD)	;PER VEDERE MEGLIO ALLO OSCILLOSCOPIO

;	XOR	B		;COMPLEMENTA ANCORA BIT RICHIESTI

;	OUT	(PIOAD),A



;	IN	A,(PIOAD)

;	AND	0F0H		;RESETTA 4 LSB

;	OR	B

;	OUT	(PIOAD),A	;EMETTI DATO VOLUTO

;	POP	BC

;	RET

;	ENDIF

	

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



;			  AREA COSTANTI ROM



;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



ROMSTR:	DB	SOH,'00','0',STX,'P',ETX,80H



TABVECS:DW	ILANTX

	DW	LANSTAT

	DW	LANRX

	DW	LANSPEC



NAKROM:	DB	NAK

ACKROM:	DB	ACK



TABSIO:	DB	18H		;CHANNEL RESET

	DB	10H		;RESET EXT/STATUS INT

	DB	0C0H		;RESET TX UNDERRUN/EOM LATCH

	DB	30H		;ERROR RESET

	

	IF	CLK16

	DB	4,045H		;CLOCK X 16 - NO SYNC - 1 STOP BIT

				;PARITY ODD

	ENDIF

	

	IF	CLK64

	DB	4,0C5H		;CLOCK X 64 - NO SYNC - 1 STOP BIT

				;PARITY ODD

	ENDIF

	

	DB	3,041H		;RX ENABLE - 7 BIT CHAR

				;

	DB	5,2AH		;TX ENABLE - 7 BIT CHAR - RTS

				;

	DB	10H		;RESET EXT/STATUS INT

	DB	1,16H		;INT ON ALL RX CHARS - PARITY DOES NOT

				;AFFECT VECTOR - INT ON TX

				;NO EXT INT ENABLE



DRVEND:



;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *



;			       AREA RAM



;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

;* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *

	



	DSEG

	

;	BUFFER TRASMISSIONE DI DEFAULT



LBUF:				;COPIA RAM DEL MESSAGGIO DEFAULT DI POLLING

LSOH:		DS	1	;SOH

LDESTH:		DS	1	;GRUPPO

LDESTL:		DS	1	;NUMERO DEL DESTINATARIO

LNMSG:		DS	1	;NUMERO DEL MESSAGGIO

LSTX:		DS	1	;STX

LCOM:		DS	1	;COMANDO

LETX:		DS	1	;ETX

LBCC:		DS	1	;BCC

ENDSTR:		DS	2	;SPAZIO IN PIU'



;	BUFFER TRASMISSIONE UTENTE



UBUF:

USOH:		DS	1	;SOH

UDESTH:		DS	1	;GRUPPO

UDESTL:		DS	1	;NUMERO DEL DESTINATARIO (ASCII)

UNMSG:		DS	1	;NUMERO DEL MESSAGGIO

USTX:		DS	1	;STX

UCOM:		DS	1	;COMANDO

		DS	LANBUFL-6	;BUFFER MESSAGGI GBASIC

UBLAST:		DS	1



;	PUNTATORI DI TRASMISSIONE



LANTXPO:	DS	2	;POINTER AL BUFFER DI TRASMISSIONE

LANWRPO:	DS	2	;POINTER DI SCRITTURA IN BUFFER UTENTE

LANTXCNT:	DS	1	;CONTATORE CARATTERI TRASMESSI LAN

GBTXCNT:	DS	1	;CONTATORE CARATTERI ACCUMULATI IN BUFFER UT.



;	BUFFER DI RICEZIONE E RELATIVI PUNTATORI



LANRXBU:			;IDENTIFICA PRIMO CARATTERE DEL BUFFER

RSOH:		DS	1	;SOH

RGA:		DS	1	;GA

RID:		DS	1	;IDENT

RNMSG:		DS	1	;NUMERO DI MESSAGGIO

RSTX:		DS	1	;STX

RKEYCH:		DS	1	;DI SOLITO CARATTERE-CHIAVE DELLA RISPOSTA

RFROM:		DS	LANBUFL-3 ;BUFFER DI RICEZIONE LAN

RBLAST:		DS	1



CNTREAD:	DS	2	;CONTATORE CARATTERI PER READBUF

READBUF:	DS	LANBUFL	;BUFFER DI TRANSITO PER FRASE READ





LANRXPO:	DS	2	;POINTER AL BUFFER DI RICEZIONE

LANRDPO:	DS	2	;POINTER DI LETTURA BUFFER

LANRXCNT:	DS	1	;CONTATORE CARATTERI RICEVUTI LAN



;	VARIABILI CHE IDENTIFICANO STATI E LORO ACCESSORI



RXSTAT:		DS	1	;STATO DELLA RICEZIONE - VEDI DESCRIZIONE

				;SUL MANUALE E EQU INIZIALI PER I VALORI

TIMSTAT:	DS	1	;STATO DEL PROCESSO TIMER - VEDI MANUALE

				;ED EQU INIZIALI

PRESC100:	DS	1	;VALORE ATTUALE PRESCALER DI PROCLAN



;	CONTATORI DI RIPETIZIONI



WAITCNT:	DS	1	;CONTATORE PERMANENZA IN STATO WAITING

REPCNT:		DS	1	;CONTATORE RITRASMISSIONI MESSAGGIO

NAKCNT:		DS	1	;CONTATORE RITRASMISSIONI NAK

MAXRTY:		DS	1	;NUMERO MASSIMO DI RETRY (NAK E RITRA)



;	VARIABILI RELATIVE A PARAMETRI DI COMUNICAZIONE



EXPNMSG:	DS	1	;NUMERO DI MESSAGGIO CHE CI SI ATTENDE IN RISPOSTA

RXERR:		DS	1	;CODICE DELL'EVENTUALE ERRORE VEDI MANUALE ED

				;EQU INIZIALI

RXBCC:		DS	1	;BCC PROGRESSIVO CALCOLATO DURANTE LA RICEZIONE

PDEST:		DS	1	;STAZIONE DI DESTINAZIONE (BINARIO)

UDEST:		DS	1	;STAZ. DI DEST. RICHIESTA DA UTENTE (BINARIO)



;	FLAGS VARI



BUF2:		DS	1	; =0 INDICA TRASMISSIONE POLLING DI DEFAULT

				;<>0 INDICA TRASMISSIONE BUFFER GBASIC

MREADY:		DS	1	;<>0 INDICA RISPOSTA IN ATTESA DI ESSERE LETTA

;	VARIABILI RELATIVE A GESTIONE STAZIONI A STAZIONI ONLINE / OFFLINE



LASTON:		DS	1	;ULTIMA STAZIONE ONLINE  INDIRIZZATA

LASTOFF:	DS	1	;ULTIMA STAZIONE OFFLINE INDIRIZZATA

ONCOUNT:	DS	1	;CONTATORE DI TRASMISSIONI EFFETTUATE A STAZIONI

				;ONLINE

WAKEUP:		DS	1	;FLAG SETTATO QUANDO SI RICEVE UN MESSAGGIO

				;DA UNA STAZIONE CHE SI RITENEVA IN OFF

STARTUP:	DS	1	;FLAG FASE STARTUP - RESTA AD UNO FINCHE'

				;NON ARRIVA IL PRIMO MESSAGGIO BASIC

				;FINO AD ALLORA LE RISPOSTE NON SONO ANALIZZATE



PRE_BUF2:	DS	1	;RICHIESTA DI ATTIVARE BUF2 AL PROSSIMO IDLE

BCCH:		DS	1	;BBCH RICEVUTO



;	ATTENZIONE - LE TRE LOC CHE SEGUONO DEVONO STARE CONTIGUE



CTLP:	DS	1		;SIO PORTA CONTROLLO

DTAP:	DS	1		;SIO PORTA DATI

VECP:	DS	1		;SIO PORTA VETTORE

CHNUM:	DS	1		;NUMERO DEL CANALE COME VISTO DAL GBASIC



;	TABELLA STAZIONI - CONTENUTI SONO QUELLI DELLA PROSSIMA TRASMISSIONE



PUMPTAB:

NMFRP0:		DS	1	;LABEL: NMSG FROM PUMP 0

NMTOP0:		DS	1	:       NMSG TO PUMP 0

PSTAT:		DS	1	;       STATUS DELLA STAZIONE ON/OFF LINE



		DS	SLAVMAX*3-3	;SPAZIO PER LE ALTRE STAZIONI





	END

